

UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

Slingram data processing and profiling programs

By

Paul V. Raab

Open-File Report 83-461

1983

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS. Programs presented in this report have been written in BASIC for a Hewlett-Packard 85 computer. Although program tests have been made no guarantee (expressed or implied) is made by the author regarding program correctness or proper execution on all computer systems.

Introduction

This report presents BASIC programs for processing and displaying Slingram type geophysical data using a Hewlett-Packard 85 desktop computer. In the Slingram method, two electromagnetic coils (a transmitter and a receiver) are separated by a fixed distance and connected with a reference cable. This coil system is moved along a traverse line and measurements are made at selected points on the line. The receiver measures the in-phase and out-of-phase components of the earth response to a time-varying magnetic field produced by the transmitter. Typically, measurements are made at several transmitted frequencies in order to characterize the electrical section at each station. The HP-85 programs presented in this report produce graphic displays of the Slingram data which can outline geophysical anomalies and aid in planning further studies. Since the HP-85 is portable, data profiles can be quickly drawn in the field.

The program "SLDATA" (Appendix I) is used to enter, edit and store Slingram type data. The programs "SLPLT1" (Appendix II) and "SLPLT2" (Appendix III) retrieve the stored data, apply corrections, and display the results in graphic and/or tabular form. "SLPLT1" uses the HP-85's CRT and thermal printer for printing and plotting. "SLPLT2" uses the HP-82905A for printed output and the HP-7225A or HP-7470A for plotting profiles. Each program is written in BASIC for a Hewlett-Packard 85 desktop computer equipped with the Plotter/Printer, Input/Output, Advanced Programming, and Matrix ROM's, and 32K bytes of memory.

Slingram Data Entry Program "SLDATA"

Slingram type data is entered and stored using the program "SLDATA". The user enters header information consisting of a project title, line number, date, coil spacing, station number multiplier, and a list of measured frequencies (a maximum of five). Slingram field data is entered for a maximum of 150 stations. Field data for each station consists of the in-phase and out-of-phase components measured for each frequency, and a correction factor. Header records and field data can be edited after keyboard entry or upon retrieval from a mass storage device. The user can print header records and field data on the HP-85's internal printer or the HP-82905A printer.

After loading and starting the program "SLDATA", all eight special function keys (SFK) of the HP-85 are labeled. Pressing one of these keys will select different program options as follows:

HELP - a brief explanation of the program will be printed or displayed.

KEYS - selects the header record and field data entry routine. The user is prompted to enter a project title, line number, date, coil spacing, and a station multiplier. A maximum length is specified for each entry. The user then enters up to 5 frequencies using a list format (i.e. frequencies are entered separated by commas). Next, field data for each station and frequency are entered in list format as follows: station number, in-phase(freq 1), out-of-phase(freq 1), in-phase(freq 2), out-of-phase(freq 2), ..., and a correction factor. This last value is used in calculating topographic corrections to the in-phase values. For example, if the transmitter and receiver coils were held coplanar and at a constant coil separation during the survey, no topographic correction need be entered. If the coil spacing is longer or shorter than the nominal spacing, then the actual coil spacing or the slope between coils in % of grade must be entered for each station. If the coils were not held coplanar but were instead held perpendicular to gravity, then the slope between coils must be entered in % of grade. Data are sorted by station number and so can be entered in any order. When field data for all stations have been entered, "END LINE" is pressed without entering a value and program control returns to the option select routine. Figure 1 illustrates a sample keyboard entry session.

READ - allows the user to recover previously stored data; usually to edit the header records and/or field data. The user must enter a data file name. Loading a new data set into memory erases all previously entered data.

EDIT - executes the editing subroutine for both header records and field data. If header records are edited, each record is displayed and the user is prompted to enter a new value for the record. If "END LINE" is pressed without entering a new value, the header record is left unchanged. Data values are edited by station number. The station data is displayed in the same list format and order used to enter data in the subroutine "KEYS". To leave the data unchanged, press "END LINE". If changes are to be made, move the cursor within the data list and edit individual values. To stop editing, press "END LINE" without entering a station number. Figure 2 describes a sample editing session.

ADD - this subroutine is used to add new stations to a data base in memory. Header records are not modified. Field data entry is the same as for "KEYS".

DEL - this subroutine is used to delete stations from a data base in memory. The user is queried for the station number to be deleted. To return to the main program, simply press "END LINE" without entering a value.

STORE - when the user is satisfied that the header records and data values are complete and correct, pressing this key executes a data storage routine. Data may be stored on either a tape cartridge or floppy disk. A name for the output file is requested. Next, one must specify whether the output file is new or old (i.e. whether there is another file with the same name on the mass storage medium being used). If it is an old file then the duplicate file will be purged and replaced. The user is queried again before a file is purged.

PRINT - executes a subroutine which will print out the header records and data values on the HP-85's internal printer or on the HP-82905A impact printer (Figure 3).

Slingram Data Processing Programs "SLPLT1" and "SLPLT2"

The programs "SLPLT1" and "SLPLT2" retrieve and process Slingram data stored by the program "SLDATA". Several corrections are applied to the raw data. An instrument error correction is applied for each individual data collection system; i.e. a zero offset is subtracted from the data and the result is normalized by a scale factor. These values are relatively constant for a particular system and so are entered into the programs via "DATA" statements. In this listing of the programs the zero offsets are set equal to "0" and the scale factors are set equal to "1" such that no instrument correction is made.

Topographic corrections to the in-phase component can be made for three cases: (1) receiver and transmitter coil separation varies from the nominal separation and the actual separation between coils is known, (2) receiver and transmitter coil separation varies from the nominal separation and the slope in % of grade between coils is known, (3) receiver and transmitter coils are not coplanar and the slope in % grade is known. — Depending on the particular survey setup, use of none, one, or two of these corrections would be desired. The equations used in the program are, respectively, as follows:

for case (1),

$$IPc = IPo + ((1 - (Sn/Sa)**3) * 100)$$

for case (2),

$$IPc = IPo + ((1 - (\cos(\arctan(\%grade/100))) ** 3) * 100)$$

for case (3),

$$IPc = IPo + (3 * (\sin(\arctan(\%grade/100)) ** 2) * 100)$$

where IPc = the corrected in-phase component, IPo = the observed in-phase component, Sn = the nominal coil spacing, Sa = the actual coil spacing, and %grade = the slope between coils in % of grade (Keller and Frischknecht, 1966, p. 381).

The program user has the option of adding 100% to all of the in-phase components. The instrument reading of some Slingram systems has had 100% subtracted from the displayed data. Finally, the in-phase component of the lowest frequency can be subtracted from the in-phase components of all the other frequencies. All corrections are made as the data are being read from tape.

Corrected data can be printed and/or plotted. The program "SLPLT1" uses the HP-85's CRT and thermal printer for output. The program "SLPLT2" uses the HP-82905A impact printer and the HP-7225A pen plotter for output. Plotting options include several profile formats and X and Y axis scaling.

After loading and starting the programs "SLPLT1" or "SLPLT2", three special function keys of the HP-85 are labeled. Pressing one of these keys will select the following program options:

DATA - enter data from a mass storage device and make corrections. The user must enter a data file name. Next, the user selects which topographic corrections are to be applied, specifies whether to add 100% to all in-phase components and specifies whether to subtract the in-phase of the lowest frequency from the in-phase components of all other frequencies. Header records and field data are then read and corrections are applied.

PRINT - prints corrected Slingram data on the selected output device. Figure 4a illustrates an HP-85 type printout and Figure 4b illustrates an HP-82905A type printout.

PLOT - plots corrected Slingram data with several options. For example, the user can select specific frequencies for plotting; profiles can be stacked in any order, plotted individually, or overlain on top of each other; the X and Y axes can be automatically scaled or the user can select minimum and maximum values for the Y-axis and a map scale

for the X-axis. If profiles are plotted at a map scale on the HP-7225A, several sheets of paper may be required. The HP-85 will beep and display the message "LOAD PAPER AND CONTINUE" if this is necessary. If profiles are plotted on the HP-85's CRT and copied to its thermal printer, a continuous profile will be created (Caution: map scales on the HP-85's CRT are not as accurate as map scales on the HP-7225A). Symbols, lines, or both symbols and lines can be used to represent data on the profiles. If lines are drawn, a bi-cubic spline function is used to smooth the profile (de Boor, 1978). The user must select a data interval for calculation of the spline. Usually, an interval equal to one fifth the station spacing is sufficient. Figure 5 describes a sample plotting session. Figures 6a, 6b, 6c, and 6d illustrate some examples of profiles generated on the HP-85's CRT and on the HP-7225A.

When any of the above subroutines have completed, the program again labels the special function keys of the HP-85. In this way, many different types of profiles can be generated from one data set at a time.

References

de Boor, Carl, 1978, A practical guide to splines: Applied Mathematical Sciences, v. 17, Springer-Verlag, New York, 392 p.

Keller, George V., and Frischknecht, Frank C., 1966, Induction methods: Electrical Methods in Geophysical Prospecting, Pergamon Press, New York, 517 p.

ENTER PROJECT TITLE - 32 max	
?	
TEST DATA - H-COPLANAR LOOPS	
ENTER LINE NUMBER - 25 max	
?	
1	
ENTER DATE - 8 max	
?	
2/22/83	
ENTER COIL SPACING - 5 max	
?	
100	
ENTER STATION MULTIPLIER - 6 max	
(e.g. IF STA 20 IS 2000 UNITS,	
THEN MULTIPLIER EQUALS 100)	
?	
1	
ENTER INPUT FREQUENCY LIST	
DEFAULT IS 322,444,888,1777,3555	
?	
322,444,888,1777,3555	
ENTER S,IP(1),OP(1),...,IP(N),	ENTER S,IP(1),OP(1),...,IP(N),
OP(N),CORR IN LIST FORMAT	OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY	HIT ONLY END LINE TO QUIT ENTRY
?	?
75,-4,0,-2,2,4,1,9,-8,9,-33	200,-5,-8,-5,-9,0,-15,-5,-31,-28
	, -55,7
ENTER S,IP(1),OP(1),...,IP(N),	ENTER S,IP(1),OP(1),...,IP(N),
OP(N),CORR IN LIST FORMAT	OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY	HIT ONLY END LINE TO QUIT ENTRY
?	?
100,-5,-2,-4,-1,0,-3,3,-20,0,-36	325,0,-5,4,-4,10,-12,7,-33,-22,-
	59,2,3
ENTER S,IP(1),OP(1),...,IP(N),	ENTER S,IP(1),OP(1),...,IP(N),
OP(N),CORR IN LIST FORMAT	OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY	HIT ONLY END LINE TO QUIT ENTRY
?	?
125,-6,-6,-6,-7,-5,-13,-10,-26,-	250,-5,-4,2,-1,11,-9,6,-33,-22,-
23,-43	55,1,7
ENTER S,IP(1),OP(1),...,IP(N),	ENTER S,IP(1),OP(1),...,IP(N),
OP(N),CORR IN LIST FORMAT	OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY	HIT ONLY END LINE TO QUIT ENTRY
?	?
150,-5,-11,-7,-16,-19,-24,-32,-3	275,-8,-1,-1,0,9,-11,2,-34,-27,-
5,-46,-45	54,6,3
ENTER S,IP(1),OP(1),...,IP(N),	ENTER S,IP(1),OP(1),...,IP(N),
OP(N),CORR IN LIST FORMAT	OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY	HIT ONLY END LINE TO QUIT ENTRY
?	?
175,-13,-13,-14,-18,-16,-25,-25,	300,-11,1,-2,3,9,-8,3,-33,-24,-5
-37,-51,-47,12,3	4
	ENTER S,IP(1),OP(1),...,IP(N),
	OP(N),CORR IN LIST FORMAT
	HIT ONLY END LINE TO QUIT ENTRY
	?
	325,-19,7,2,10,17,-2,15,-28,-12,
	-54
	ENTER S,IP(1),OP(1),...,IP(N),
	OP(N),CORR IN LIST FORMAT
	HIT ONLY END LINE TO QUIT ENTRY
	?
	350,-9,9,5,13,26,1,25,-27,-4,-55
	ENTER S,IP(1),OP(1),...,IP(N),
	OP(N),CORR IN LIST FORMAT
	HIT ONLY END LINE TO QUIT ENTRY
	?
	SORTING DATA - PLEASE WAIT
	WAITING FOR KEY SELECTION

Figure 1. Sample of keyboard entry of data using the program "SLDATA".


```

TEST DATA - H-COPLANER LOOPS
LINE = 1
2/22/83
COIL SPACING = 100
MULTIPLY STATION * 1

EDIT HEADER? Y/N - DEF IS NO
?
Y

TEST DATA - H-COPLANER LOOPS
ENTER NEW PROJECT TITLE - 32 max
HIT ONLY END LINE TO KEEP SAME
?

LINE = 1
ENTER NEW LINE NUMBER - 25 max
HIT ONLY END LINE TO KEEP SAME
?

2/22/83
ENTER NEW DATE - 8 max
HIT ONLY END LINE TO KEEP SAME
?
8/17/83

COIL SPACING = 100
ENTER NEW COIL SPACING - 5 max
HIT ONLY END LINE TO KEEP SAME
?

MULTIPLY STATION * 1
ENTER STATION MULTIPLIER - 5 max
(e.g. IF STA 20 IS 2000 FEET,
THEN MULTIPLIER EQUALS 100)
HIT ONLY END LINE TO KEEP SAME
?

```

```

EDIT DATA? Y/N - DEF IS NO
?
Y
ENTER STATION NUMBER TO EDIT
HIT ONLY END LINE TO CONTINUE
?
75

ENTER S,IP(1),OP(1), ...,IP(N),
OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY
?
75,-4.0,-2.2,4.1,9,-8.9,-33.0

ENTER STATION NUMBER TO EDIT
HIT ONLY END LINE TO CONTINUE
?
200

ENTER S,IP(1),OP(1), ...,IP(N),
OP(N),CORR IN LIST FORMAT
HIT ONLY END LINE TO QUIT ENTRY
?
200,-5,-8,-5,-9.0,-15,-5,-31,-28
,-55.7

ENTER STATION NUMBER TO EDIT
HIT ONLY END LINE TO CONTINUE
?

SORTING DATA - PLEASE WAIT
WAITING FOR KEY SELECTION

```

Figure 2. Sample of data and header editing session using the program "SLDATA".

INPUT FILE NAME = TEST

SLINGRAM DATA

TEST DATA - H-COPLANER LOOPS LINE = 1 8/17/83

COIL SPACING = 100 MULTIPLY STATION * 1

NUMBER OF FREQUENCIES = 5 NUMBER OF STATIONS = 12

```

*****
* FREQUENCY * FREQUENCY * FREQUENCY * FREQUENCY * FREQUENCY * CORR *
* 222 * 444 * 888 * 1777 * 3555 * *
*****
STATION IP OP IP OP IP OP IP OP IP OP
+75.0 -4.0 +0.0 -2.0 +2.0 +4.0 +1.0 +9.0 -8.0 +9.0 -33.0 +0.0
+100.0 -8.0 -2.0 -4.0 -1.0 +0.0 -3.0 +3.0 -20.0 +0.0 -36.0 +0.0
+125.0 -6.0 -6.0 -6.0 -7.0 -5.0 -13.0 -10.0 -26.0 -23.0 -43.0 +0.0
+150.0 -5.0 -11.0 -7.0 -16.0 -19.0 -24.0 -22.0 -35.0 -46.0 -49.0 +0.0
+175.0 -13.0 -13.0 -14.0 -18.0 -16.0 -25.0 -28.0 -51.0 -47.0 +12.0
+200.0 -5.0 -8.0 -9.0 -9.0 +0.0 -15.0 -15.0 -31.0 -28.0 -49.0 +7.0
+225.0 +0.0 -3.0 +4.0 -4.0 +10.0 -12.0 +7.0 -13.0 -22.0 -19.0 +2.0
+250.0 -8.0 -4.0 +2.0 -1.0 +11.0 -9.0 +6.0 -13.0 -22.0 -19.0 +1.0
+275.0 -8.0 -1.0 -1.0 +0.0 +9.0 -11.0 +3.0 -14.0 -27.0 -24.0 +6.0
+300.0 -11.0 +1.0 -3.0 +3.0 +9.0 -8.0 +6.0 -13.0 -27.0 -24.0 +6.0
+325.0 -10.0 +7.0 +2.0 +10.0 +17.0 -2.0 +15.0 -28.0 -12.0 -54.0 +0.0
+350.0 -9.0 +9.0 +5.0 +13.0 +26.0 +1.0 +25.0 -27.0 -4.0 -55.0 +0.0

```

Figure 3. Printed output of Slingram data on the HP-82905A impact printer using the program "SLDATA".

INPUT FILE NAME = TEST
 SLINGRAM DATA
 TEST DATA - H-COPLANER LOOPS
 LINE = 1
 8/17/93
 COIL SPACING = 100
 MULTIPLY STATION * 1
 NUMBER OF FREQUENCIES = 5
 NUMBER OF STATIONS = 12

FREQUENCY 888

STATION	IP	OP
+75.0	+104.0	+1.0
+100.0	+100.0	-3.0
+125.0	+95.0	-13.0
+150.0	+81.0	-24.0
+175.0	+86.2	-25.0
+200.0	+100.7	-15.0
+225.0	+110.1	-12.0
+250.0	+111.0	-9.0
+275.0	+109.6	-11.0
+300.0	+109.0	-8.0
+325.0	+117.0	-2.0
+350.0	+126.0	+1.0

FREQUENCY 222

STATION	IP	OP
+75.0	+96.0	+0.0
+100.0	+95.0	-2.0
+125.0	+94.0	-6.0
+150.0	+95.0	-11.0
+175.0	+89.2	-13.0
+200.0	+95.7	-8.0
+225.0	+100.1	-5.0
+250.0	+95.0	-4.0
+275.0	+92.6	-1.0
+300.0	+89.0	+1.0
+325.0	+90.0	+7.0
+350.0	+91.0	+9.0

FREQUENCY 1777

STATION	IP	OP
+75.0	+109.0	-8.0
+100.0	+103.0	-20.0
+125.0	+90.0	-26.0
+150.0	+78.0	-35.0
+175.0	+77.2	-37.0
+200.0	+95.7	-31.0
+225.0	+107.1	-33.0
+250.0	+106.0	-33.0
+275.0	+102.6	-34.0
+300.0	+103.0	-33.0
+325.0	+115.0	-28.0
+350.0	+125.0	-27.0

FREQUENCY 444

STATION	IP	OP
+75.0	+98.0	+2.0
+100.0	+96.0	-1.0
+125.0	+94.0	-7.0
+150.0	+93.0	-16.0
+175.0	+88.2	-18.0
+200.0	+95.7	-9.0
+225.0	+104.1	-4.0
+250.0	+102.0	-1.0
+275.0	+99.6	+0.0
+300.0	+98.0	+3.0
+325.0	+102.0	+10.0
+350.0	+105.0	+13.0

FREQUENCY 3555

STATION	IP	OP
+75.0	+109.0	-33.0
+100.0	+100.0	-36.0
+125.0	+77.0	-43.0
+150.0	+54.0	-45.0
+175.0	+51.2	-47.0
+200.0	+72.7	-55.0
+225.0	+78.1	-59.0
+250.0	+78.0	-55.0
+275.0	+73.6	-54.0
+300.0	+76.0	-54.0
+325.0	+88.0	-54.0
+350.0	+96.0	-55.0

Figure 4a. "SLPLT1" program printed output of corrected Slingram data on the HP-85's internal printer.

INPUT FILE NAME = TEST

SLINGRAM DATA

TEST DATA - H-COPLANER LOOPS LINE = 1 8/17/83

COIL SPACING = 100 MULTIPLY STATION * 1

NUMBER OF FREQUENCIES = 5 NUMBER OF STATIONS = 12

```

*****
*   FREQUENCY   *   FREQUENCY   *   FREQUENCY   *   FREQUENCY   *   FREQUENCY   *
*       222     *       444     *       888     *       1777    *       3555    *
*****
STATION  IP      OP      IP      OP      IP      OP      IP      OP      IP      OP
+75.0    +96.0    +0.0    +98.0    +2.0    +104.0    +1.0    +109.0    -8.0    +109.0    -33.0
+100.0    +95.0    -2.0    +96.0    -1.0    +100.0    -3.0    +103.0    -20.0    +100.0    -36.0
+125.0    +94.0    -6.0    +94.0    -7.0    +95.0    -13.0    +90.0    -26.0    +77.0    -43.0
+150.0    +95.0    -11.0    +93.0    -16.0    +81.0    -24.0    +78.0    -35.0    +54.0    -45.0
+175.0    +89.2    -13.0    +88.2    -18.0    +86.2    -25.0    +77.2    -37.0    +51.2    -47.0
+200.0    +95.7    -8.0    +95.7    -9.0    +100.7    -15.0    +95.7    -31.0    +72.7    -55.0
+225.0    +100.1    -5.0    +104.1    -4.0    +110.1    -12.0    +107.1    -33.0    +78.1    -59.0
+250.0    +95.0    -4.0    +102.0    -1.0    +111.0    -9.0    +108.0    -33.0    +78.0    -55.0
+275.0    +92.6    -1.0    +99.6    +0.0    +109.6    -11.0    +102.6    -34.0    +73.6    -54.0
+300.0    +89.0    +1.0    +98.0    +3.0    +109.0    -8.0    +103.0    -33.0    +76.0    -54.0
+325.0    +90.0    +7.0    +102.0    +10.0    +117.0    -2.0    +115.0    -28.0    +88.0    -54.0
+350.0    +91.0    +9.0    +105.0    +13.0    +126.0    +1.0    +125.0    -27.0    +96.0    -55.0

```

Figure 4b. "SLPLT2" printed output of corrected Slingram data on the HP-82905A impact printer.

ENTER PLOTTER ADDRESS
DEFAULT=708
?

DEFAULT PLOT TITLE IS
TEST DATA - H-COPLANER LOOPS/LIN
E = 1/COIL SPACING = 100
ENTER NEW TITLE TO CHANGE(80max)
ELSE HIT ONLY END LINE
?

ENTER LIST OF FREQUENCIES
TO PLOT. DEFAULT IS ALL
?

SELECT FORMAT OPTION
1=STACKED PROFILES
2=OVERLAY PROFILES
3=INDIVIDUAL PROFILES
DEFAULT=1
?

SELECT Y-AXIS SCALING OPTION
1=GROUP MAX-MIN AUTO SCALING
2=IND MAX-MIN AUTO SCALING
3=USER SELECTED SCALES
DEFAULT=1
?

SELECT X-AXIS INTERCEPT OPTION
1=CENTER OF Y-AXIS
2=FLOATING INTERCEPT
DEFAULT=1
?

SELECT X-AXIS SCALING OPTION
1=MAX-MIN AUTO SCALING
2=USER SELECTED MAP SCALES
(E.G. 1:62500)
DEFAULT=1
?

DISTANCE IN FEET(1) OR METERS(2)
DEFAULT=1
?

SELECT PLOTTING OPTION
1=VECTORS
2=SYMBOLS
3=BOTH
DEFAULT=3
?

ENTER SMOOTHING INTERVAL
?
5
CALCULATING MAX/MIN VALUES

Figure 5. Sample of a plotting session from the program "SLPLT2". Results of the selected plotting options (all defaults) are shown in Figure 6c.




Figure 6a. Stacked profiles of Slingram data plotted on the HP-85.




Figure 6b. A single Slingram frequency plotted on the HP-85.

TEST DATA - H-COPLANER LOOPS/LINE = 1/COIL SPACING = 100
 IP = SOLID LINE AND (*) OP = DASHED LINE AND (+)




Figure 6c. Stacked Slingram profiles plotted on the HP-7225A using the default plotting options of program "SLPLT2". 100% added to the in-phase component.

TEST DATA - H-COPLANER LOOPS/LINE = 1/COIL SPACING = 100
 IP = SOLID LINE OP = DASHED LINE




Figure 6d. Overlain Slingram profiles plotted on the HP-7225A.

Appendix I. Program listing for "SLDATA".

```

10 ! ***** SLDATA *****
20 ! * PROGRAM TO ENTER, EDIT,*
30 ! * AND STORE SLINGRAM TYPE*
40 ! * EM DATA. HEADER INFO  *
50 ! * AND FIELD DATA ARE    *
60 ! * PROCESSED. USER CAN    *
70 ! * PRINT DATA ON THE HP-85*
80 ! * OR HP-82905A.          *
90 ! *****
100 ! BY PAUL RAAB - USGS
110 ! MODIFIED 10/30/83
120 OPTION BASE 1
130 CRT IS 1 @ PRINTER IS 2
140 SHORT F(5),S(150),R(150,5),Q(150,5),T(150)
150 DIM
A$(32),B$(32),C$(8),D$(20),E$(25),F$(25),G$(32),H$(25)
160 DIM I$(96),N$(17),O$(76),T$(80),TO$(12),TI$(30),Z$(64)
170 IOBUFFER TI$
180 MAT F=ZER@ MAT S=ZER@ MAT R=ZER@ MAT Q=ZER@ MAT T=ZER
190 !
200 ! OPTION DRIVER
210 !
220 ON KEY# 1,"HELP" GOSUB 4350
230 ON KEY# 2,"KEYS" GOSUB 340
240 ON KEY# 3,"READ" GOSUB 740
250 ON KEY# 4,"EDIT" GOSUB 1110
260 ON KEY# 5,"ADD" GOSUB 1840
270 ON KEY# 6,"DEL" GOSUB 2040
280 ON KEY# 7,"STORE" GOSUB 2310
290 ON KEY# 8,"PRINT" GOSUB 2710
300 CLEAR @ SFLAG 1 @ KEY LABEL
310 DISP "WAITING FOR KEY SELECTION"
320 IF FLAG(1) THEN 320 ELSE 300 ! WAIT FOR KEY HIT
330 !
340 ! KEYBOARD DATA ENTRY
350 !
360 CLEAR @ CFLAG 1
370 N$=""
380 DISP "ENTER PROJECT TITLE - 32 max" @ INPUT A$
390 DISP @ DISP "ENTER LINE NUMBER - 25 max"
400 B$(1,7)="LINE = " @ INPUT B$(8)
410 DISP @ DISP "ENTER DATE - 8 max" @ INPUT C$
420 DISP @ DISP "ENTER COIL SPACING - 5 max"
430 D$(1,15)="COIL SPACING = " @ INPUT D$(16)
440 DISP @ DISP "ENTER STATION MULTIPLIER - 6 max"
450 DISP "(e.g. IF STA 20 IS 2000 UNITS,"
460 DISP "THEN MULTIPLIER EQUALS 100)"
470 E$(1,19)="MULTIPLY STATION * " @ INPUT E$(20)

```

```

480 MAT F=ZER@ MAT S=ZER@ MAT R=ZER@ MAT Q=ZER@ MAT T=ZER
490 DISP @ DISP "ENTER INPUT FREQUENCY LIST"
500 LINPUT "DEFAULT IS 222,444,888,1777,3555?",G$
510 IF G$="" THEN G$="222,444,888,1777,3555"
520 M,P1=1
530 P2=POS(G$[P1],",")+P1-2
540 IF P2>=P1 THEN F(M)=VAL(G$[P1,P2])
550 IF P2<P1 THEN F(M)=VAL(G$[P1]) @ GOTO 580
560 P1=P2+2 @ M=M+1
570 GOTO 530
580 F$="NUMBER OF FREQUENCIES = "&VAL$(M)
590 N=0
600 FOR I=1 TO 150
610 DISP @ DISP "ENTER S,IP(1),OP(1),...,IP(N),"
620 DISP "OP(N),CORR IN LIST FORMAT"
630 LINPUT "HIT ONLY END LINE TO QUIT ENTRY ?",I$
640 IF I$="" THEN 700
650 CFLAG 3
660 GOSUB 3660 ! DATA COLLECT
670 IF FLAG(3) THEN 610
680 N=N+1
690 NEXT I
700 H$="NUMBER OF STATIONS = "&VAL$(N)
710 GOSUB 3990 ! SORT
720 RETURN
730 !
740 ! MASS STORAGE ENTRY
750 !
760 CLEAR @ CFLAG 1
770 ON ERROR GOTO 3900
780 DISP "ENTER INPUT FILE NAME" @ INPUT N$
790 ASSIGN# 1 TO N$
800 CRT OFF
810 MAT F=ZER@ MAT S=ZER@ MAT R=ZER@ MAT Q=ZER@ MAT T=ZER
820 READ# 1 ; A$,B$,C$,D$,E$,F$,G$,H$
830 M=VAL(F$[25])
840 P1=1
850 FOR I=1 TO M
860 P2=POS(G$[P1],",")+P1-2
870 IF P2>=P1 THEN F(I)=VAL(G$[P1,P2]) ELSE F(I)=VAL(G$[P1])
880 P1=P2+2
890 NEXT I
900 N=VAL(H$[22])
910 FOR I=1 TO N
920 O$=""
930 READ# 1 ; O$
940 S(I)=VAL(O$[1,10])
950 P1=11
960 FOR J=1 TO M
970 P2=P1+5
980 R(I,J)=VAL(O$[P1,P2])

```

```

990 P1=P2+1
1000 P2=P1+5
1010 Q(I,J)=VAL(O$(P1,P2))
1020 P1=P2+1
1030 NEXT J
1040 T(I)=VAL(O$(P1))
1050 NEXT I
1060 ASSIGN# 1 TO *
1070 CRT ON
1080 OFF ERROR
1090 RETURN
1100 !
1110 ! EDIT HEADER AND DATA
1120 !
1130 CLEAR @ CFLAG 1
1140 DISP A$ @ DISP B$ @ DISP C$ @ DISP D$ @ DISP E$
1150 DISP @ DISP "EDIT HEADER? Y/N - DEF IS NO" @ INPUT Z$
1160 IF Z$="" THEN 1460
1170 IF UPC$(Z$[1,1])="N" THEN 1460
1180 IF UPC$(Z$[1,1])#"Y" THEN 1150
1190 DISP @ DISP A$
1200 DISP "ENTER NEW PROJECT TITLE - 32 max"
1210 DISP "HIT ONLY END LINE TO KEEP SAME" @ INPUT Z$
1220 IF Z$="" THEN 1240
1230 IF LEN(Z$)<=32 THEN A$[1,32]=Z$ ELSE A$[1,32]=Z$[1,32]
1240 DISP @ DISP B$
1250 DISP "ENTER NEW LINE NUMBER - 25 max"
1260 DISP "HIT ONLY END LINE TO KEEP SAME" @ INPUT Z$
1270 IF Z$="" THEN 1290
1280 IF LEN(Z$)<=25 THEN B$[8,32]=Z$ ELSE B$[8,32]=Z$[1,25]
1290 DISP @ DISP C$
1300 DISP "ENTER NEW DATE - 8 max"
1310 DISP "HIT ONLY END LINE TO KEEP SAME" @ INPUT Z$
1320 IF Z$="" THEN 1340
1330 IF LEN(Z$)<=8 THEN C$[1,8]=Z$ ELSE C$[1,8]=Z$[1,8]
1340 DISP @ DISP D$
1350 DISP "ENTER NEW COIL SPACING - 5 max"
1360 DISP "HIT ONLY END LINE TO KEEP SAME" @ INPUT Z$
1370 IF Z$="" THEN 1390
1380 IF LEN(Z$)<=5 THEN D$[16,20]=Z$ ELSE D$[16,20]=Z$[1,5]
1390 DISP @ DISP E$
1400 DISP "ENTER STATION MULTIPLIER - 6 max"
1410 DISP "(e.g. IF STA 20 IS 2000 FEET,"
1420 DISP "THEN MULTIPLIER EQUALS 100)"
1430 DISP "HIT ONLY END LINE TO KEEP SAME" @ INPUT Z$
1440 IF Z$="" THEN 1460
1450 IF LEN(Z$)<=6 THEN E$[20,25]=Z$ ELSE E$[20,25]=Z$[1,6]
1460 DISP @ DISP "EDIT DATA? Y/N - DEF IS NO" @ INPUT Z$
1470 IF Z$="" THEN RETURN
1480 IF UPC$(Z$[1,1])="N" THEN RETURN
1490 CLEAR

```

```

1500 DISP "ENTER STATION NUMBER TO EDIT"
1510 DISP "HIT ONLY END LINE TO CONTINUE" @ INPUT Z$
1520 IF Z$="" THEN 1800
1530 OFF CURSOR
1540 K=VAL(Z$[1])
1550 FOR I=1 TO N
1560 IF S(I)=K THEN 1590
1570 NEXT I
1580 DISP @ DISP "NO SUCH STATION NUMBER!" @ DISP @ GOTO
1500
1590 CLEAR @ OFF CURSOR @ ALPHA 0
1600 I$=VAL$(S(I))&","
1610 FOR J=1 TO M
1620 I$=I$&VAL$(R(I,J))&","&VAL$(Q(I,J))&","
1630 NEXT J
1640 I$=I$&VAL$(T(I))
1650 L=CEIL(LEN(I$)/32)
1660 L1=CURSROW+L
1670 IF L1>64 THEN L1=L1-64
1680 L2=L1-4
1690 IF L2<1 THEN L2=L2+64
1700 ALPHA L1 @ DISP I$ @ OFF CURSOR
1710 ALPHA L2
1720 DISP "ENTER S,IP(1),OP(1),...,IP(N),"
1730 DISP "OP(N),CORR IN LIST FORMAT"
1740 LINPUT "HIT ONLY END LINE TO QUIT ENTRY ?",I$
1750 IF I$="" THEN 1800
1760 CFLAG 3
1770 GOSUB 3660 ! DATA COLLECT
1780 IF FLAG(3) THEN 1590
1790 GOTO 1490 ! RETURN
1800 GOSUB 3990 ! SORT
1810 ON CURSOR
1820 RETURN
1830 !
1840 ! ADD STATIONS TO DATA
1850 !
1860 CLEAR @ CFLAG 1
1870 N1=0
1880 FOR I=N+1 TO 100
1890 DISP "ENTER S,IP(1),OP(1),...,IP(N),"
1900 DISP "OP(N),CORR IN LIST FORMAT"
1910 LINPUT "HIT ONLY END LINE TO QUIT ENTRY ?",I$
1920 IF I$="" THEN 1990
1930 CFLAG 3
1940 GOSUB 3660 ! DATA COLLECT
1950 IF FLAG(3) THEN DISP @ GOTO 1890
1960 N1=N1+1
1970 DISP
1980 NEXT I
1990 N=N+N1

```

```

2000 H$="NUMBER OF STATIONS = "&VAL$(N)
2010 GOSUB 3990 ! SORT
2020 RETURN
2030 !
2040 ! DELETE STATIONS
2050 !
2060 CLEAR @ CFLAG 1
2070 DISP "ENTER STATION NUMBER TO DELETE"
2080 DISP "TO RETURN, HIT ONLY END LINE" @ INPUT Z$
2090 IF Z$="" THEN 2270
2100 S=VAL(Z$[1])
2110 FOR I=1 TO N
2120 IF S(I)=S THEN 2150
2130 NEXT I
2140 DISP @ DISP "STATION NOT FOUND" @ DISP @ GOTO 2070
2150 DISP @ DISP "DELETING STATION NUMBER ";S @ DISP
2160 FOR J=I TO N
2170 IF J=N THEN 2250
2180 S(J)=S(J+1)
2190 T(J)=T(J+1)
2200 FOR K=1 TO M
2210 R(J,K)=R(J+1,K)
2220 Q(J,K)=Q(J+1,K)
2230 NEXT K
2240 NEXT J
2250 N=N-1
2260 GOTO 2070
2270 H$[22,25]=VAL$(N)
2280 GOSUB 3990 ! SORT
2290 RETURN
2300 !
2310 ! DATA STORAGE SUBROUTINE
2320 !
2330 CLEAR @ CFLAG 1
2340 ON ERROR GOTO 3900
2350 DISP "ENTER OUTPUT FILE NAME" @ INPUT N$
2360 DISP @ DISP "NEW OR OLD NAME? N/O"
2370 DISP "IF OLD, FILE WILL BE REPLACED"
2380 INPUT Z$
2390 IF UPC$(Z$[1,1])="N" THEN 2440
2400 IF UPC$(Z$[1,1])#"O" THEN 2360
2410 DISP @ DISP "PURGE FILE ";N$;"? Y/N" @ INPUT Z$
2420 IF UPC$(Z$[1,1])="Y" THEN PURGE N$
2430 IF UPC$(Z$[1,1])#"Y" THEN DISP @ GOTO 2350
2440 NO=CEIL((N*79+203)/256)*3
2450 NO=CEIL((N*79+203+NO)/256)
2460 CREATE N$,NO,256
2470 ASSIGN# 1 TO N$
2480 CRT OFF
2490 PRINT# 1 ; A$,B$,C$,D$,E$,F$,G$,H$
2500 FOR I=1 TO N

```

```

2510 O$=""
2520 O$[1,10]=VAL$(S(I))
2530 P1=11
2540 FOR J=1 TO M
2550 P2=P1+5
2560 O$[P1,P2]=VAL$(R(I,J))
2570 P1=P2+1
2580 P2=P1+5
2590 O$[P1,P2]=VAL$(Q(I,J))
2600 P1=P2+1
2610 NEXT J
2620 P2=P1+5
2630 O$[P1,P2]=VAL$(T(I))
2640 PRINT# 1 ; O$
2650 NEXT I
2660 ASSIGN# 1 TO *
2670 CRT ON
2680 OFF ERROR
2690 RETURN
2700 !
2710 ! DATA PRINT ROUTINE
2720 !
2730 CLEAR @ CFLAG 1
2740 DISP "ENTER PRINTER ADDRESS"
2750 DISP "CRT(1), PRINTER(2), 82905A(701)" @ INPUT Z$
2760 IF Z$="" THEN P=2 @ GOTO 2780
2770 P=INT(VAL(Z$))
2780 IF P>2 THEN GOTO 3050
2790 !
2800 ! HP-85 PRINT
2810 !
2820 PRINTER IS P
2830 IF P=2 THEN CRT OFF
2840 PRINT @ PRINT
2850 IF LEN(N$)#0 THEN PRINT "INPUT FILE NAME = ";N$
2860 PRINT @ PRINT "SLINGRAM DATA"
2870 PRINT A$ @ PRINT B$ @ PRINT C$ @ PRINT D$
2880 PRINT E$ @ PRINT F$ @ PRINT H$
2890 PRINT @ PRINT
2900 FOR J=1 TO M
2910 PRINT "FREQUENCY ";F(J)
2920 PRINT @ PRINT " STATION CORR IP OP"
2930 PRINT "-----"
2940 FOR I=1 TO N
2950 PRINT USING 2960 ; S(I),T(I),R(I,J),Q(I,J)
2960 IMAGE 1X,S5D.D,1X,S3D.D,2(1X,S3D.D)
2970 NEXT I
2980 PRINT @ PRINT
2990 NEXT J
3000 PRINT @ PRINT @ PRINT @ PRINT
3010 IF P=2 THEN CRT ON

```

```

3020 PRINTER IS 2
3030 RETURN
3040 !
3050 ! IMPACT PRINTER ROUTINE
3060 !
3070 CLEAR @ CRT OFF
3080 PRINTER IS P,80
3090 OUTPUT P USING "#,K" ; CHR$(27)&"C"&CHR$(72)
3100 OUTPUT P USING "#,K" ; CHR$(27)&"C"&CHR$(0)&CHR$(11)
3110 OUTPUT P USING "#,K" ; CHR$(27)&"O"
3120 OUTPUT P USING "#,K" ; CHR$(27)&"N"&CHR$(16)
3130 IF LEN(N$)#0 THEN PRINT "INPUT FILE NAME = ";N$
3140 PRINT @ PRINT "SLINGRAM DATA"
3150 PRINT TRIM$(A$)&" "&TRIM$(B$)&" "&TRIM$(C$)
3160 PRINT TRIM$(D$)&" "&TRIM$(E$) @ PRINT TRIM$(F$)&"
"&TRIM$(H$) @ PRINT
3170 T$="&RPT$("*****",M)&"*****"
3180 PRINT T$
3190 T$="&RPT$(" ",M)&" "
3200 PRINT T$
3210 T$="&RPT$(" FREQUENCY ",M)&" CORR "
3220 PRINT T$
3230 T$=""
3240 FOR I=1 TO M
3250 TO$=VAL$(F(I))
3260 L1=FLOOR((12-LEN(TO$))/2)
3270 L2=CEIL((12-LEN(TO$))/2)-1
3280 T$=T$&RPT$(" ",L1)&TO$&RPT$(" ",L2)&"*"
3290 NEXT I
3300 T$="&T$&" "
3310 PRINT T$
3320 T$="&RPT$(" ",M)&" "
3330 PRINT T$
3340 T$="*****"&RPT$("*****",M)&"*****"
3350 PRINT T$
3360 T$="&RPT$(" ! ",M)&" !"
3370 PRINT T$
3380 T$=" STATION !"&RPT$(" IP ! OP !",M)&" !"
3390 PRINT T$
3400 T$="&RPT$(" ! ",M)&" !"
3410 PRINT T$
3420 T$="!!!!!!!!!"&RPT$("!!!!!!!!!",M)&"!!!!!!!!!"
3430 PRINT T$
3440 FOR I=1 TO N
3450 T1$=""
3460 OUTPUT T1$ USING 3470 ; S(I)
3470 IMAGE #,1X,S5D.D,"!"
3480 T$=T1$
3490 FOR J=1 TO M
3500 T1$=""
3510 OUTPUT T1$ USING 3520 ; R(I,J),Q(I,J)

```

```

3520 IMAGE #,S2D.D,"!",S2D.D,"!"
3530 T$=T$&T1$
3540 NEXT J
3550 T1$=""
3560 OUTPUT T1$ USING 3570 ; T(I)
3570 IMAGE #,1X,S2D.D,1X,"!"
3580 T$=T$&T1$
3590 PRINT T$
3600 NEXT I
3610 PRINT CHR$(12) ! FORM-FEED
3620 PRINTER IS 2
3630 CRT ON
3640 RETURN
3650 !
3660 ! DATA READ (LIST FORMAT)
3670 !
3680 ON ERROR GOTO 3850
3690 CFLAG 2
3700 L=LEN(I$)
3710 P1=1
3720 P2=POS(I$[P1],",")+P1-2
3730 S(I)=VAL(I$[P1,P2])
3740 FOR J=1 TO M
3750 P1=P2+2
3760 P2=POS(I$[P1],",")+P1-2
3770 R(I,J)=VAL(I$[P1,P2])
3780 P1=P2+2
3790 P2=POS(I$[P1],",")+P1-2
3800 IF P2>=P1 THEN Q(I,J)=VAL(I$[P1,P2])
3810 IF P2<P1 THEN Q(I,J)=VAL(I$[P1])
3820 NEXT J
3830 IF P2<P1 THEN T(I)=0 ELSE T(I)=VAL(I$[P2+2])
3840 GOTO 3870
3850 SFLAG 3
3860 BEEP @ DISP @ DISP "TRY AGAIN" @ WAIT 1000
3870 OFF ERROR
3880 RETURN
3890 !
3900 ! TAPE ERROR ROUTINE
3910 !
3920 OFF ERROR @ CRT ON
3930 DISP @ DISP "ERROR NUMBER ";ERKN
3940 DISP "ERROR LINE ";ERRL @ ERRM
3950 BEEP @ DISP @ DISP "TAPE ACTION ABORTED"
3960 DISP "CONTINUE WHEN READY" @ PAUSE
3970 GOTO 200
3980 !
3990 ! SHELL-METZNER SORT
4000 !
4010 CLEAR
4020 DISP "SORTING DATA - PLEASE WAIT"

```



```

4030 T=N
4040 T=INT(T/2)
4050 IF T=0 THEN 4330
4060 T1=N-T
4070 T2=1
4080 T3=T2
4090 T4=T3+T
4100 IF S(T3)>S(T4) THEN 4140
4110 T2=T2+1
4120 IF T2>T1 THEN 4040
4130 GOTO 4080
4140 T5=S(T3)
4150 Z5=T(T3)
4160 FOR J=1 TO M
4170 R5(J)=R(T3,J)
4180 Q5(J)=Q(T3,J)
4190 NEXT J
4200 S(T3)=S(T4)
4210 T(T3)=T(T4)
4220 FOR J=1 TO M
4230 R(T3,J)=R(T4,J)
4240 R(T4,J)=R5(J)
4250 Q(T3,J)=Q(T4,J)
4260 Q(T4,J)=Q5(J)
4270 NEXT J
4280 S(T4)=T5
4290 T(T4)=Z5
4300 T3=T3-T
4310 IF T3<1 THEN 4110
4320 GOTO 4090
4330 RETURN
4340 !
4350 ! HELP ROUTINE
4360 !
4370 CLEAR @ CFLAG 1
4380 DISP "PRINT OR DISP INFORMATION? P/D"
4390 DISP "DEFAULT IS DISP" @ INPUT Z$
4400 IF Z$="" THEN Z$="D" @ GOTO 4420
4410 IF UPC$(Z$[1,1])="P" THEN CRT IS 2 ELSE CRT IS 1
4420 CLEAR @ KEY LABEL
4430 DISP "THE SPECIAL FUNCTION KEYS ARE"
4440 DISP "LABELED AT THE BOTTOM OF THE"
4450 DISP "HP-85 CRT SCREEN. UPPER LABELS"
4460 DISP "REFER TO SHIFTED KEYS AND"
4470 DISP "LOWER LABELS REFER TO UNSHIFTED"
4480 DISP "KEYS. EACH KEY, WHEN PRESSED,"
4490 DISP "WILL PERFORM THE FOLLOWING" DEFINED
FUNCTIONS."
4500 DISP
4510 IF UPC$(Z$[1,1])#"P" THEN DISP "(PRESS CONTINUE)" @
PAUSE

```

```

4520 CLEAR @ KEY LABEL
4530 DISP "HELP - AN ON LINE HELP LIBRARY"
4540 DISP "EXPLAINING PROGRAM OPTIONS."
4550 DISP
4560 DISP "KEYS - SUBROUTINE TO ENTER"
4570 DISP "HEADER INFORMATION AND DATA"
4580 DISP "FROM THE KEYBOARD."
4590 DISP
4600 DISP "READ - SUBROUTINE TO RECOVER"
4610 DISP "DATA PREVIOUSLY STORED ON TAPE."
4620 DISP
4630 IF UPC$(Z$(1,1))#"P" THEN DISP "(PRESS CONTINUE)" @
PAUSE
4640 CLEAR @ KEY LABEL
4650 DISP "EDIT - SUBROUTINE TO EDIT"
4660 DISP "HEADER, STATION ID, AND DATA."
4670 DISP "OBSERVATIONS ARE SELECTED FOR"
4680 DISP "EDITING BY STATION NUMBER."
4690 DISP
4700 DISP "ADD - ADDS NEW STATIONS TO DATA"
4710 DISP "SETS ENTERED FROM THE KEYBOARD"
4720 DISP "OR RECOVERED FROM TAPE FILES."
4730 DISP
4740 IF UPC$(Z$(1,1))#"P" THEN DISP "(PRESS CONTINUE)" @
PAUSE
4750 CLEAR @ KEY LABEL
4760 DISP "DEL - DELETES SELECTED STATIONS FROM DATA SETS."
4770 DISP
4780 DISP "STORE - SUBROUTINE TO STORE"
4790 DISP "DATA ON TAPE. USER CAN PURGE"
4800 DISP "AND RE-STORE DATA FILES WHICH HAVE BEEN
EDITED."
4810 DISP
4820 DISP "PRINT - PRINTS HEADER AND DATA"
4830 DISP "ON THE HP-85 OR HP-82905A."
4840 DISP
4850 IF UPC$(Z$(1,1))#"P" THEN DISP "(PRESS CONTINUE)" @
PAUSE
4860 CRT IS 1
4870 RETURN

```

Appendix II. Program listing for "SLPLT1".

```

10 ! ***** SLPLT1 *****
20 ! * SLINGRAM DATA PLOTTING *
30 ! * PROGRAM. PRINTING AND *
40 ! * PLOTTING OF CORRECTED *
50 ! * SLINGRAM DATA ON THE *
60 ! * HP-85'S PRINTER AND *
70 ! * PLOTTER. OPTIONS FOR *
80 ! * USER SELECTED SCALES. *
90 ! *****
100 ! BY PAUL RAAB - USGS
110 ! MODIFIED 10/30/83
120 OPTION BASE 1
130 CRT IS 1 @ PRINTER IS 2
140 DEG
150 SHORT F(5),E0(5),E1(5),E2(5),E3(5),E4(5),E5(5)
160 SHORT S(150),R(150,5),R1(5),R2(5)
170 SHORT Q(150,5),Q1(5),Q2(5),G1(5),G2(5)
180 SHORT A(150),B(150),C(150)
190 DIM
A$(32),B$(32),C$(8),D$(20),E$(25),F$(25),G$(32),H$(25)
200 DIM T$(80),T0$(12),T1$(30),T2$(75),T3$(75)
210 DIM N$(17),O$(76),Z$(32)
220 IOBUFFER T1$
230 ! IN-PHASE SCALE FACTOR
240 DATA 1,1,1,1,1
250 ! IN-PHASE ZERO OFF <.2
260 DATA 0,0,0,0,0
270 ! IN-PHASE ZERO OFF >.2
280 DATA 0,0,0,0,0
290 MAT READ E0@ MAT READ E1@ MAT READ E2
300 ! OUT-OF-PHASE SCALE FACTOR
310 DATA 1,1,1,1,1
320 ! OUT-OF-PHASE ZERO OFF <.2
330 DATA 0,0,0,0,0
340 ! OUT-OF-PHASE ZERO OFF >.2
350 DATA 0,0,0,0,0
360 MAT READ E3@ MAT READ E4@ MAT READ E5
370 !
380 ! OPTION SELECT
390 !
400 ON KEY# 1,"DATA" GOSUB 480
410 ON KEY# 2,"PRINT" GOSUB 1260
420 ON KEY# 3,"PLOT" GOSUB 1570
430 CLEAR @ SFLAG 1 @ KEY LABEL
440 DISP "WAITING FOR KEY SELECTION"
450 IF FLAG(1) THEN 450 ELSE 430 ! WAIT FOR KEY HIT
460 END
470 !

```

```

480 ! READ TAPE
490 !
500 CLEAR @ CFLAG 1
510 DISP "ENTER INPUT FILE NAME" @ INPUT N$
520 DISP @ DISP "SELECT IN-PHASE TOPO CORRECTION"
530 DISP "0=NO CORRECTIONS"
540 DISP "1=COIL SPACING CORRECTION (DIST)"
550 DISP "2=COIL SPACING CORRECTION (%GR)"
560 DISP "3=COIL TILT CORRECTION (%GR)"
570 DISP "4=BOTH CORRECTIONS (%GR)"          DEFAULT=0"
580 INPUT Z$
590 IF Z$="" THEN F0=0 @ GOTO 610
600 F0=VAL(Z$)
610 DISP @ DISP "ADD 100% TO IN-PHASE? Y/N"
620 DISP "DEFAULT IS YES" @ INPUT Z$
630 IF Z$="" THEN SFLAG 2 @ GOTO 650
640 IF UPC$(Z$[1,1])="N" THEN CFLAG 2 ELSE SFLAG 2
650 DISP @ DISP "SUBTRACT LOW FREQUENCY IN-PHASE"
660 DISP "FROM OTHER FREQUENCIES? Y/N"
670 DISP "DEFAULT IS NO" @ INPUT Z$
680 IF Z$="" THEN CFLAG 3 @ GOTO 700
690 IF UPC$(Z$[1,1])="Y" THEN SFLAG 3 ELSE CFLAG 3
700 ASSIGN# 1 TO N$
710 CRT OFF
720 READ# 1 ; A$,B$,C$,D$,E$,F$,G$,H$
730 C=VAL(D$[16])
740 S0=VAL(E$[20])
750 M=VAL(F$[25])
760 P1=1
770 F=1.E99
780 FOR I=1 TO M
790 P2=POS(G$[P1],",")+P1-2
800 IF P2>=P1 THEN F(I)=VAL(G$[P1,P2]) ELSE F(I)=VAL(G$[P1])
810 P1=P2+2
820 F=MIN(F,F(I))
830 NEXT I
840 P0=11+M*12
850 N=VAL(H$[22])
860 FOR I=1 TO N
870 O$=""
880 READ# 1 ; O$
890 S(I)=VAL(O$[1,10])
900 T=VAL(O$[P0])
910 IF F0=1 AND T>0 THEN C1=(1-(C/T)*3)*100 ELSE C1=0
920 IF F0=2 OR F0=4 THEN C2=(1-COS(ATN(T/100))*3)*100
930 IF F0=3 OR F0=4 THEN C3=SIN(ATN(T/100)) @ C3=3*C3*C3*100
940 P1=11
950 FOR J=1 TO M
960 P2=P1+5
970 R(I,J)=VAL(O$[P1,P2])
980 IF ABS(R(I,J))<20 THEN R(I,J)=(R(I,J)+E1(J))/E0(J) @

```

```

GOTO 1000
990 IF ABS(R(I,J))>=20 THEN R(I,J)=(R(I,J)+E2(J))/E0(J)
1000 IF F0=1 THEN R(I,J)=R(I,J)+C1
1010 IF F0=2 OR F0=4 THEN R(I,J)=R(I,J)+C2
1020 IF F0=3 OR F0=4 THEN R(I,J)=R(I,J)+C3
1030 IF FLAG(2) THEN R(I,J)=R(I,J)+100
1040 P1=P2+1
1050 P2=P1+5
1060 Q(I,J)=VAL(O$[P1,P2])
1070 IF ABS(Q(I,J))<20 THEN Q(I,J)=(Q(I,J)+E4(J))/E3(J) @
GOTO 1090
1080 IF ABS(Q(I,J))>=20 THEN Q(I,J)=(Q(I,J)+E5(J))/E3(J)
1090 P1=P2+1
1100 NEXT J
1110 NEXT I
1120 ASSIGN# 1 TO *
1130 S1=S(1) @ S2=S(N)
1140 IF NOT FLAG(3) THEN 1240
1150 FOR K=1 TO M
1160 IF F(K)=F THEN 1180
1170 NEXT K
1180 FOR I=1 TO N
1190 FOR J=1 TO M
1200 IF J=K THEN 1220
1210 R(I,J)=R(I,J)-R(I,K)
1220 NEXT J
1230 NEXT I
1240 RETURN
1250 !
1260 ! HP-85 PRINT
1270 !
1280 CLEAR @ CFLAG 1
1290 DISP "ENTER PRINTER ADDRESS"
1300 DISP "CRT(1) OR PRINTER(2) - DEF=1" @ INPUT Z$
1310 IF Z$="" THEN P=1 @ GOTO 1340
1320 P=INT(VAL(Z$))
1330 IF P>2 THEN DISP "NOT AN HP-85 DEVICE" @ DISP @ GOTO
1290
1340 PRINTER IS P
1350 IF P=2 THEN CRT OFF
1360 PRINT @ PRINT
1370 PRINT "INPUT FILE NAME = ";N$
1380 PRINT @ PRINT "SLINGRAM DATA"
1390 PRINT A$ @ PRINT B$ @ PRINT C$ @ PRINT D$
1400 PRINT E$ @ PRINT F$ @ PRINT H$
1410 PRINT @ PRINT
1420 FOR J=1 TO M
1430 PRINT "FREQUENCY ";F(J)
1440 PRINT @ PRINT " STATION IP OP"
1450 PRINT "-----"
1460 FOR I=1 TO N

```

```

1470 PRINT USING 1480 ; S(I),R(I,J),Q(I,J)
1480 IMAGE 1X,S5D.D,2(2X,S3D.D)
1490 NEXT I
1500 PRINT @ PRINT
1510 NEXT J
1520 PRINT @ PRINT @ PRINT @ PRINT
1530 IF P=2 THEN CRT ON
1540 PRINTER IS 2
1550 RETURN
1560 !
1570 ! PLOTTING ROUTINE
1580 !
1590 CLEAR @ CFLAG 1
1600 DISP "ENTER PLOTTER NUMBER"
1610 DISP "CRT(1) OR PRINTER(2)"           DEFAULT=2"
1620 INPUT Z$
1630 IF Z$="" THEN P=2 @ GOTO 1650
1640 IF VAL(Z$)=1 THEN P=1 ELSE P=2
1650 DISP
1660 DISP "ENTER FREQUENCY LIST"
1670 LINPUT "DEFAULT IS ALL"               ?",Z$
1680 IF Z$="" THEN Z$=G$
1690 P1,M0=1
1700 FOR I=1 TO M
1710 P2=POS(Z$[P1],",")+P1-2
1720 IF P2>=P1 THEN F0(I)=VAL(Z$[P1,P2])
1730 IF P2<P1 THEN F0(I)=VAL(Z$[P1]) @ GOTO 1770
1740 M0=M0+1
1750 P1=P2+2
1760 NEXT I
1770 DISP
1780 DISP "SELECT FORMAT OPTION"
1790 DISP "1=STACKED PROFILES"             2=OVERLAY
PROFILES"
1800 DISP "3=INDIVIDUAL PROFILES"         DEFAULT=3"
1810 INPUT Z$
1820 IF Z$="" THEN F1=3 ELSE F1=VAL(Z$[1])
1830 DISP
1840 DISP "SELECT Y-AXIS SCALING OPTION"
1850 DISP "1=GROUP MAX-MIN AUTO SCALING"
1860 DISP "2=IND MAX-MIN AUTO SCALING"
1870 DISP "3=USER SELECTED SCALES"
1880 DISP "DEFAULT=1"
1890 INPUT Z$
1900 IF Z$="" THEN F2=1 ELSE F2=VAL(Z$[1])
1910 IF F1=2 AND F2=2 THEN F2=1 @ GOTO 1990
1920 IF F2#3 THEN 1990
1930 IF FLAG(2) THEN DISP @ DISP "ENTER IN-PHASE MIN,MAX"
1940 INPUT R1,R2
1950 IF FLAG(2) THEN DISP @ DISP "ENTER OUT-OF-PHASE
MIN,MAX"

```

```

1960 INPUT Q1,Q2
1970 IF NOT FLAG(2) THEN DISP @ DISP "ENTER GROUP MIN,MAX"
1980 INPUT G1,G2
1990 DISP
2000 DISP "SELECT X-AXIS INTERCEPT OPTION 1=CENTER OF
Y-AXIS"
2010 DISP "2=FLOATING INTERCEPT DEFAULT=1"
2020 INPUT Z$
2030 IF Z$="" THEN F3=1 ELSE F3=VAL(Z$[1])
2040 DISP
2050 DISP "SELECT X-AXIS SCALING OPTION"
2060 DISP "1=MAX-MIN AUTO SCALING"
2070 DISP "2=USER SELECTED MAP SCALES (E.G. 1:62500)"
2080 DISP "DEFAULT=1"
2090 INPUT Z$
2100 IF Z$="" THEN F4=1 ELSE F4=VAL(Z$[1])
2110 IF F4=1 THEN 2140
2120 IF F4=2 THEN DISP @ DISP "ENTER MAP SCALE SELECTED"
2130 INPUT U3
2140 DISP @ DISP "STATION'S IN FEET OR METERS? F/M"
2150 DISP "DEFAULT='F'" @ INPUT Z$
2160 IF Z$="" THEN F5=1 @ GOTO 2180
2170 IF UPC$(Z$[1,1])="M" THEN F5=2 ELSE F5=1
2180 DISP
2190 DISP "SELECT PLOTTING OPTION 1=VECTORS"
2200 DISP "2=SYMBOLS 3=BOTH"
2210 DISP "DEFAULT=3"
2220 INPUT Z$
2230 IF Z$="" THEN F6=3 ELSE F6=VAL(Z$[1])
2240 DISP
2250 IF F6=1 OR F6=3 THEN DISP "ENTER SMOOTHING INTERVAL"
2260 INPUT S3@ S3=S3/S0
2270 IF FLAG(2) THEN Y0=100 ELSE Y0=0
2280 !
2290 ! Y-AXIS MAX/MIN ROUTINE
2300 !
2310 IF F2=3 THEN 2970
2320 CLEAR @ DISP "CALCULATING MAX/MIN VALUES"
2330 G1=1.E99 @ G2=-1.E99
2340 R1=1.E99 @ R2=-1.E99 @ Q1=1.E99 @ Q2=-1.E99
2350 FOR K=1 TO M0
2360 G1(K)=1.E99 @ G2(K)=-1.E99
2370 R1(K)=1.E99 @ R2(K)=-1.E99 @ Q1(K)=1.E99 @ Q2(K)=-1.E99
2380 FOR J=1 TO M
2390 IF F0(K)=F(J) THEN 2410
2400 NEXT J
2410 FOR I=1 TO N
2420 R1(K)=MIN(R1(K),R(I,J))
2430 R2(K)=MAX(R2(K),R(I,J))
2440 Q1(K)=MIN(Q1(K),Q(I,J))
2450 Q2(K)=MAX(Q2(K),Q(I,J))

```

```

2460 NEXT I
2470 R1(K)=FLOOR(R1(K)/10)*10
2480 R2(K)=CEIL(R2(K)/10)*10
2490 IF FLAG(2) THEN 2530
2500 IF R1(K)>=0 THEN R1(K)=-10
2510 IF R2(K)<=0 THEN R2(K)=10
2520 GOTO 2570
2530 IF R1(K)>=100 THEN R1(K)=90
2540 IF R2(K)<=100 THEN R2(K)=110
2550 IF F3=1 AND R2(K)-100>100-R1(K) THEN R1(K)=200-R2(K)
2560 IF F3=1 AND 100-R1(K)>R2(K)-100 THEN R2(K)=200-R1(K)
2570 Q1(K)=FLOOR(Q1(K)/10)*10
2580 Q2(K)=CEIL(Q2(K)/10)*10
2590 IF Q1(K)>=0 THEN Q1(K)=-10
2600 IF Q2(K)<=0 THEN Q2(K)=10
2610 IF FLAG(2) THEN 2670
2620 G1(K)=MIN(R1(K),Q1(K))
2630 G2(K)=MAX(R2(K),Q2(K))
2640 IF F3=1 AND G2(K)>-G1(K) THEN G1(K)=-G2(K)
2650 IF F3=1 AND -G1(K)>G2(K) THEN G2(K)=-G1(K)
2660 G1=MIN(G1,G1(K)) @ G2=MAX(G2,G2(K))
2670 R1=MIN(R1,R1(K)) @ R2=MAX(R2,R2(K))
2680 Q1=MIN(Q1,Q1(K)) @ Q2=MAX(Q2,Q2(K))
2690 NEXT K
2700 !
2710 ! CALCULATE OP MAX/MIN
2720 !
2730 IF NOT FLAG(2) THEN 2970
2740 IF F2=1 THEN 2870
2750 FOR K=1 TO M0
2760 IF Q1(K)>R1(K)-100 THEN Q1(K)=R1(K)-100
2770 IF Q1(K)<R1(K)-100 THEN R1(K)=Q1(K)+100
2780 IF Q2(K)<R2(K)-100 THEN Q2(K)=R2(K)-100
2790 IF Q2(K)>R2(K)-100 THEN R2(K)=Q2(K)+100
2800 IF F3#1 THEN 2850
2810 IF R2(K)-100>100-R1(K) THEN R1(K)=200-R2(K)
2820 IF R2(K)-100<100-R1(K) THEN R2(K)=200-R1(K)
2830 IF Q2(K)>-Q1(K) THEN Q1(K)=-Q2(K)
2840 IF Q2(K)<-Q1(K) THEN Q2(K)=-Q1(K)
2850 NEXT K
2860 GOTO 2970
2870 IF Q1>R1-100 THEN Q1=R1-100
2880 IF Q1<R1-100 THEN R1=Q1+100
2890 IF Q2<R2-100 THEN Q2=R2-100
2900 IF Q2>R2-100 THEN R2=Q2+100
2910 IF F3#1 THEN 2970
2920 IF R2-100>100-R1 THEN R1=200-R2
2930 IF R2-100<100-R1 THEN R2=200-R1
2940 IF Q2>-Q1 THEN Q1=-Q2
2950 IF Q2<-Q1 THEN Q2=-Q1
2960 !

```



```

2970 ! X-AXIS MIN/MAX ROUTINE
2980 !
2990 X1=S1
3000 IF F4=1 THEN X2=S2
3010 IF F4=2 AND F5=1 THEN X2=X1+89.8*U3/304.8/S0
3020 IF F4=2 AND F5=2 THEN X2=X1+89.8*U3/1000/S0
3030 X3=1
3040 IF X2-X1>=4 AND X2-X1<10 THEN X3=2
3050 IF X2-X1>=10 AND X2-X1<25 THEN X3=5
3060 IF X2-X1>=25 AND X2-X1<50 THEN X3=10
3070 IF X2-X1>=50 AND X2-X1<100 THEN X3=25
3080 IF X2-X1>=100 AND X2-X1<2500 THEN X3=50
3090 IF X2-X1>=250 AND X2-X1<5000 THEN X3=100
3100 IF X2-X1>=500 AND X2-X1<10000 THEN X3=250
3110 IF X2-X1>=1000 AND X2-X1<2500 THEN X3=500
3120 IF X2-X1>=2500 AND X2-X1<50000 THEN X3=1000
3130 IF X2-X1>=5000 AND X2-X1<100000 THEN X3=2500
3140 IF X2-X1>=10000 THEN X3=5000
3150 X1=FLOOR(X1/X3)*X3
3160 IF F4=2 AND F5=1 THEN X2=X1+89.8*U3/304.8/S0
3170 IF F4=2 AND F5=2 THEN X2=X1+89.8*U3/1000/S0
3180 V1=1 ! INITIAL PAPER COUNT
3190 !
3200 ! PLOT SETUP
3210 !
3220 CLEAR @ GCLEAR
3230 PLOTTER IS 1
3240 LIMIT 0,97,0,72
3250 LOCATE 0,RATIO*100,0,100
3260 IF F1=1 THEN D0=85/M0 ELSE D0=85
3270 FOR K=1 TO M0
3280 FOR J=1 TO M
3290 IF F(J)=F0(K) THEN 3320
3300 NEXT J
3310 GOTO 5130
3320 IF F1<3 AND K>1 THEN 3550
3330 !
3340 ! DRAFT TITLES
3350 !
3360 IF V1>1 THEN GOTO 3550
3370 IF P=1 THEN 3490
3380 PRINT @ PRINT
3390 IF F6=1 THEN PRINT "IN-PHASE = SOLID LINE"
3400 IF F6=1 THEN PRINT "OUT-PHASE = SOLID LINE"
3410 IF F6=2 THEN PRINT "IN-PHASE = (*)"
3420 IF F6=2 THEN PRINT "OUT-PHASE = (+)"
3430 IF F6=3 THEN PRINT "IN-PHASE = SOLID LINE AND (*)"
3440 IF F6=3 THEN PRINT "OUT-PHASE = SOLID LINE AND (+)"
3450 IF F4=2 THEN PRINT "MAP SCALE = 1:"&VAL$(U3)
3460 PRINT @ PRINT
3470 IF FLAG(2) THEN PRINT "          (% ) IN-PHASE"

```

```

3480 IF NOT FLAG(2) THEN PRINT "                PERCENT (%)"
3490 SETGU
3500 MOVE RATIO*100/2,0 @ LDIR 0 @ LORG 4
3510 IF F5=1 THEN LABEL "DISTANCE ( x "&VAL$(S0)&" ft )"
3520 IF F5=2 THEN LABEL "DISTANCE ( x "&VAL$(S0)&" m )"
3530 SETUU
3540 !
3550 ! LOCATE PLOTTING AREA
3560 !
3570 IF F1>1 THEN 3610
3580 IF V1=1 THEN LOCATE 10,RATIO*100,(K-1)*D0+10,10+K*D0-5
3590 IF V1>1 THEN LOCATE 0,RATIO*100,(K-1)*D0+10,10+K*D0-5
3600 GOTO 3640
3610 IF V1=1 THEN LOCATE 10,RATIO*100,10,95
3620 IF V1>1 THEN LOCATE 0,RATIO*100,10,95
3630 !
3640 ! SCALE IN-PHASE PLOT AREA
3650 !
3660 IF F2#2 AND FLAG(2) THEN Y1=R1 @ Y2=R2
3670 IF F2#2 AND NOT FLAG(2) THEN Y1=G1 @ Y2=G2
3680 IF F2=2 AND FLAG(2) THEN Y1=R1(K) @ Y2=R2(K)
3690 IF F2=2 AND NOT FLAG(2) THEN Y1=G1(K) @ Y2=G2(K)
3700 SCALE X1,X2,Y1,Y2
3710 !
3720 ! LABEL EACH FREQUENCY
3730 !
3740 IF F1=2 THEN 3830
3750 SETGU
3760 LDIR 0 @ LORG 4
3770 IF F1=1 THEN MOVE RATIO*100/2,10+K*D0-5
3780 IF F1#1 THEN MOVE RATIO*100/2,95
3790 LABEL F(J);" Hz"
3800 PENUP
3810 SETUU
3820 !
3830 ! DRAFT TRAVERSE AXIS
3840 !
3850 LINETYPE 1
3860 IF F1=2 AND K>1 THEN 4210
3870 MOVE X1,Y0 @ DRAW X2,Y0 @ PENUP
3880 L1=CEIL(X1/X3)*X3
3890 L2=FLOOR(X2/X3)*X3
3900 FOR I=L1 TO L2 STEP X3
3910 MOVE I,Y0 @ SETGU
3920 IMOVE 0,-1 @ IDRAW 0,2
3930 PENUP @ SETUU
3940 NEXT I
3950 IF F1#3 AND K>1 THEN 4040
3960 LORG 4 @ LDIR 0
3970 IF V1=1 THEN L1=CEIL(X1/X3)*X3+X3
3980 IF V1>1 THEN L1=CEIL(X1/X3)*X3

```

```

3990 L2=FLOOR(X2/X3)*X3
4000 FOR I=L1 TO L2 STEP X3
4010 MOVE I,Y1 @ SETGU @ IMOVE 0,-4 @ LABEL I @ SETUU
4020 NEXT I
4030 !
4040 ! DRAFT IN-PHASE AXIS
4050 !
4060 IF V1>1 THEN 4210
4070 YAXIS X1,10,Y1,Y2
4080 LDIR 90 @ LORG 4
4090 IF F1=1 THEN MOVE X1,Y0 @ SETGU
4100 IF F1=1 THEN IMOVE -2,0 @ LABEL Y0 @ SETUU @ GOTO 4210
4110 L1=CEIL(Y1/10)*10
4120 L2=FLOOR(Y2/10)*10
4130 IF (Y2-Y1)/10<=4 THEN Y3=10
4140 IF (Y2-Y1)/10>4 AND (Y2-Y1)/10<=8 THEN Y3=20
4150 IF (Y2-Y1)/10>8 THEN Y3=40
4160 IF RMD(L1,Y3)#0 THEN L1=L1+ABS(RMD(L1,Y3))
4170 FOR I=L1 TO L2 STEP Y3
4180 MOVE X1,I @ SETGU @ IMOVE -2,0 @ LABEL I @ SETUU
4190 NEXT I
4200 !
4210 ! IN-PHASE LINE PLOT
4220 !
4230 F9=0
4240 IF F6=2 THEN 4390
4250 GOSUB 5600 ! SPLINE
4260 FOR I=1 TO N
4270 IF I<N THEN S4=(S(I+1)-S(I))/S3
4280 IF I=N THEN S4=(S(I)-S(I-1))/S3
4290 FOR IO=1 TO S4
4300 Z0=S(I)+(IO-1)*S3
4310 IF Z0<S1 OR Z0>S2 THEN 4360
4320 Z1=Z0-S(I)
4330 Z2=R(I,J)+Z1*(A(I)+Z1*(B(I)+Z1*C(I)))
4340 PLOT Z0,Z2
4350 NEXT IO
4360 NEXT I
4370 PENUP
4380 !
4390 ! IN-PHASE SYMBOL PLOT
4400 !
4410 IF F6=1 THEN 4480
4420 LORG 5 @ LDIR 0
4430 FOR I=1 TO N
4440 IF S(I)<X1 OR S(I)>X2 THEN 4460
4450 MOVE S(I),R(I,J) @ LABEL "*"
4460 NEXT I
4470 !
4480 ! SCALE OUT-OF-PHASE AREA
4490 !

```

```

4500 F9=1
4510 IF NOT FLAG(2) THEN 4580
4520 IF F2#2 THEN Y1=Q1 @ Y2=Q2
4530 IF F2=2 THEN Y1=Q1(K) @ Y2=Q2(K)
4540 SCALE X1,X2,Y1,Y2
4550 IF F1=2 AND K>1 OR X2<S2 THEN 4580
4560 YAXIS X2,10,Y1,Y2
4570 !
4580 ! OUT-OF-PHASE LINE PLOT
4590 !
4600 LINETYPE 1
4610 IF F6=2 THEN 4760
4620 GOSUB 5600 ! SPLINE
4630 FOR I=1 TO N
4640 IF I<N THEN S4=(S(I+1)-S(I))/S3
4650 IF I=N THEN S4=(S(I)-S(I-1))/S3
4660 FOR IO=1 TO S4
4670 Z0=S(I)+(IO-1)*S3
4680 IF Z0<S1 OR Z0>S2 THEN 4730
4690 Z1=Z0-S(I)
4700 Z2=Q(I,J)+Z1*(A(I)+Z1*(B(I)+Z1*C(I)))
4710 PLOT Z0,Z2
4720 NEXT IO
4730 NEXT I
4740 PENUP
4750 !
4760 ! OUT-OF-PHASE SYMBOL PLOT
4770 !
4780 IF F6=1 THEN 4850
4790 LORG 5 @ LDIR 0
4800 FOR I=1 TO N
4810 IF S(I)<X1 OR S(I)>X2 THEN 4830
4820 MOVE S(I),Q(I,J) @ LABEL "+"
4830 NEXT I
4840 !
4850 ! END FOR INDIVIDUALS
4860 !
4870 IF F1#3 THEN 5130
4880 IF P=2 THEN COPY
4890 GCLEAR
4900 IF X2<S2 THEN 5100
4910 IF P=1 OR NOT FLAG(2) THEN 5060
4920 LOCATE 0,RATIO*100,10,95
4930 SCALE X1,X2,Y1,Y2
4940 LDIR 270 @ LORG 4
4950 L1=CEIL(Y1/10)*10
4960 L2=FLOOR(Y2/10)*10
4970 IF (Y2-Y1)/10<=4 THEN Y3=10
4980 IF (Y2-Y1)/10>4 AND (Y2-Y1)/10<=8 THEN Y3=20
4990 IF (Y2-Y1)/10>8 THEN Y3=40
5000 IF RMD(L1,Y3)#0 THEN L1=L1+ABS(RMD(L1,Y3))

```

```

5010 FOR I=L1 TO L2 STEP Y3
5020 MOVE X1,I @ SETGU @ IMOVE 2,0 @ LABEL I @ SETUU
5030 NEXT I
5040 MOVE X1,0 @ SETGU @ IMOVE 10,0
5050 LABEL "(%) OUT-OF-PHASE" @ SETUU @ COPY @ GCLEAR
5060 V1=1 @ X1=FLOOR(S1/X3)*X3
5070 IF F4=1 THEN X2=S2 @ GOTO 5130
5080 IF F4=2 AND F5=1 THEN X2=X1+89.8*U3/304.8/S0 @ GOTO
5130
5090 IF F4=2 AND F5=2 THEN X2=X1+89.8*U3/1000/S0 @ GOTO 5130
5100 V1=V1+1 @ X1=X2
5110 IF F5=1 THEN X2=X1+97*U3/304.8/S0 @ GOTO 3550
5120 IF F5=2 THEN X2=X1+97*U3/1000/S0 @ GOTO 3550
5130 NEXT K
5140 IF F1=3 THEN RETURN
5150 IF P=2 THEN COPY
5160 GCLEAR
5170 IF X2<S2 THEN 5530
5180 IF P=1 OR NOT FLAG(2) THEN RETURN
5190 !
5200 ! END OVERLAY PROFILES
5210 !
5220 IF F1=1 THEN 5390
5230 LOCATE 0,RATIO*100,10,95
5240 SCALE X1,X2,Y1,Y2
5250 LDIR 270 @ LORG 4
5260 L1=CEIL(Y1/10)*10
5270 L2=FLOOR(Y2/10)*10
5280 IF (Y2-Y1)/10<=4 THEN Y3=10
5290 IF (Y2-Y1)/10>4 AND (Y2-Y1)/10<=8 THEN Y3=20
5300 IF (Y2-Y1)/10>8 THEN Y3=40
5310 IF RMD(L1,Y3)#0 THEN L1=L1+ABS(RMD(L1,Y3))
5320 FOR I=L1 TO L2 STEP Y3
5330 MOVE X1,I @ SETGU @ IMOVE 2,0 @ LABEL I @ SETUU
5340 NEXT I
5350 MOVE X1,0 @ SETGU @ IMOVE 10,0
5360 LABEL "(%) OUT-OF-PHASE" @ SETUU @ COPY @ GCLEAR
5370 RETURN
5380 !
5390 ! END STACKED PROFILES
5400 !
5410 LDIR 270 @ LORG 4
5420 FOR K=1 TO M0
5430 LOCATE 0,RATIO*100,(K-1)*D0+10,10+K*D0-5
5440 IF F2#2 THEN Y1=Q1 @ Y2=Q2
5450 IF F2=2 THEN Y1=Q1(K) @ Y2=Q2(K)
5460 SCALE X1,X2,Y1,Y2
5470 MOVE X1,0 @ SETGU @ IMOVE 2,0 @ LABEL "0" @ SETUU
5480 NEXT K
5490 SETGU @ MOVE 10,50
5500 LABEL "(%) OUT-OF-PHASE" @ SETUU @ COPY @ GCLEAR

```

```

5510 RETURN
5520 !
5530 ! RESET VARIABLES
5540 !
5550 V1=V1+1 @ X1=X2
5560 IF X2<S2 AND F5=1 THEN X2=X1+97*U3/304.8/S0 @ GOTO 3200
5570 IF X2<S2 AND F5=2 THEN X2=X1+97*U3/1000/S0 @ GOTO 3200
5580 RETURN
5590 !
5600 ! SPLINE SUBROUTINE
5610 !
5620 IF N<2 THEN RETURN
5630 IF N<3 THEN 6050
5640 N2=N-1
5650 C(1)=S(2)-S(1)
5660 IF F9=0 THEN B(2)=(R(2,J)-R(1,J))/C(1)
5670 IF F9=1 THEN B(2)=(Q(2,J)-Q(1,J))/C(1)
5680 FOR N4=2 TO N2
5690 C(N4)=S(N4+1)-S(N4)
5700 A(N4)=2*(C(N4-1)+C(N4))
5710 IF F9=0 THEN B(N4+1)=(R(N4+1,J)-R(N4,J))/C(N4)
5720 IF F9=1 THEN B(N4+1)=(Q(N4+1,J)-Q(N4,J))/C(N4)
5730 B(N4)=B(N4+1)-B(N4)
5740 NEXT N4
5750 A(1)=-C(1)
5760 A(N)=-C(N2)
5770 B(1)=0
5780 B(N)=0
5790 IF N=3 THEN 5840
5800 B(1)=B(3)/(S(4)-S(2))-B(2)/(S(3)-S(1))
5810 B(N)=B(N2)/(S(N)-S(N-2))-B(N-2)/(S(N2)-S(N-3))
5820 B(1)=B(1)*C(1)2/(S(4)-S(1))
5830 B(N)=-(B(N)*C(N-1)2/(S(N)-S(N-3)))
5840 FOR W=2 TO N
5850 Z0=C(W-1)/A(W-1)
5860 A(W)=A(W)-Z0*C(W-1)
5870 B(W)=B(W)-Z0*B(W-1)
5880 NEXT W
5890 B(N)=B(N)/A(N)
5900 FOR N3=1 TO N2
5910 W=N-N3
5920 B(W)=(B(W)-C(W)*B(W+1))/A(W)
5930 NEXT N3
5940 IF F9=0 THEN
A(N)=(R(N,J)-R(N2,J))/C(N2)+C(N2)*(B(N2)+2*B(N))
5950 IF F9=1 THEN
A(N)=(Q(N,J)-Q(N2,J))/C(N2)+C(N2)*(B(N2)+2*B(N))
5960 FOR W=1 TO N2
5970 IF F9=0 THEN
A(W)=(R(W+1,J)-R(W,J))/C(W)-C(W)*(B(W+1)+2*B(W))
5980 IF F9=1 THEN

```

```

A(W)=(Q(W+1,J)-Q(W,J))/C(W)-C(W)*(B(W+1)+2*B(W))
5990 C(W)=(B(W+1)-B(W))/C(W)
6000 B(W)=3*B(W)
6010 NEXT W
6020 B(N)=3*B(N)
6030 C(N)=C(N-1)
6040 GOTO 6080
6050 IF F9=0 THEN A(1)=(R(2,J)-R(1,J))/(S(2)-S(1))
6060 IF F9=1 THEN A(1)=(Q(2,J)-Q(1,J))/(S(2)-S(1))
6070 B(1)=0 @ C(1)=0 @ A(2)=A(1) @ B(2)=0 @ C(2)=0
6080 RETURN

```

Appendix III. Program listing for "SLPLT2".

```
10 ! ***** SLPLT2 *****
20 ! * SLINGRAM DATA PLOTTING *
30 ! * PROGRAM. PRINTS ON THE *
40 ! * HP-82905A AND PLOTS ON *
50 ! * THE HP-7225A. PROGRAM *
60 ! * OPTIONS FOR MAP SCALES *
70 ! * OR AUTO-SCALING. INPUT *
80 ! * FROM "SLDATA" FILES. *
90 ! *****
100 ! BY PAUL RAAB - USGS
110 ! MODIFIED 11/4/83
120 OPTION BASE 1
130 CRT IS 1 @ PRINTER IS 2
140 DEG
150 SHORT F(5),E0(5),E1(5),E2(5),E3(5),E4(5),E5(5)
160 SHORT S(150),R(150,5),R1(5),R2(5)
170 SHORT Q(150,5),Q1(5),Q2(5),G1(5),G2(5)
180 SHORT A(150),B(150),C(150)
190 DIM
A$(32),B$(32),C$(8),D$(20),E$(25),F$(25),G$(32),H$(25)
200 DIM T$(80),T0$(12),T1$(30),T2$(86),T3$(80)
210 DIM N$(17),O$(76),Z$(32)
220 IOBUFFER T1$
230 ! IN-PHASE SCALE FACTOR
240 DATA 1,1,1,1,1
250 ! IN-PHASE ZERO OFF <.2
260 DATA 0,0,0,0,0
270 ! IN-PHASE ZERO OFF >.2
280 DATA 0,0,0,0,0
290 MAT READ E0@ MAT READ E1@ MAT READ E2
300 ! OUT-OF-PHASE SCALE FACTOR
310 DATA 1,1,1,1,1
320 ! OUT-OF-PHASE ZERO OFF <.2
330 DATA 0,0,0,0,0
340 ! OUT-OF-PHASE ZERO OFF >.2
350 DATA 0,0,0,0,0
360 MAT READ E3@ MAT READ E4@ MAT READ E5
370 !
380 ! OPTION SELECT
390 !
400 ON KEY# 1,"DATA" GOSUB 480
410 ON KEY# 2,"PRINT" GOSUB 1270
420 ON KEY# 3,"PLOT" GOSUB 1960
430 CLEAR @ SFLAG 1 @ KEY LABEL
440 DISP "WAITING FOR KEY SELECTION"
450 IF FLAG(1) THEN 450 ELSE 430 ! WAIT FOR KEY HIT
460 END
470 !
```



```

480 ! READ TAPE
490 !
500 CLEAR @ CFLAG 1
510 DISP "ENTER INPUT FILE NAME" @ INPUT N$
520 DISP @ DISP "SELECT IN-PHASE CORRECTION TYPE"
530 DISP "0=NO CORRECTION"
540 DISP "1=COIL SPACING CORRECTION (DIST)"
550 DISP "2=COIL SPACING CORRECTION (%GR)"
560 DISP "3=COIL TILT CORRECTION (%GR)"
570 DISP "4=BOTH CORRECTIONS (%GR)"
580 DISP "DEFAULT=0"
590 INPUT Z$
600 IF Z$="" THEN FO=0 @ GOTO 620
610 FO=VAL(Z$)
620 DISP @ DISP "ADD 100% TO IN-PHASE? Y/N"
630 DISP "DEFAULT IS YES" @ INPUT Z$
640 IF Z$="" THEN SFLAG 2 @ GOTO 660
650 IF UPC$(Z$[1,1])="N" THEN CFLAG 2 ELSE SFLAG 2
660 DISP @ DISP "SUBTRACT LOW FREQUENCY IN-PHASE"
670 DISP "FROM OTHER FREQUENCIES? Y/N"
680 DISP "DEFAULT IS NO" @ INPUT Z$
690 IF Z$="" THEN CFLAG 3 @ GOTO 710
700 IF UPC$(Z$[1,1])="Y" THEN SFLAG 3 ELSE CFLAG 3
710 ASSIGN# 1 TO N$
720 CRT OFF
730 READ# 1 ; A$,B$,C$,D$,E$,F$,G$,H$
740 C=VAL(D$[16])
750 SO=VAL(E$[20])
760 M=VAL(F$[25])
770 P1=1
780 F=1.E99
790 FOR I=1 TO M
800 P2=POS(G$[P1],",")+P1-2
810 IF P2>=P1 THEN F(I)=VAL(G$[P1,P2]) ELSE F(I)=VAL(G$[P1])
820 P1=P2+2
830 F=MIN(F,F(I))
840 NEXT I
850 PO=11+M*12
860 N=VAL(H$[22])
870 FOR I=1 TO N
880 O$=""
890 READ# 1 ; O$
900 S(I)=VAL(O$[1,10])
910 T=VAL(O$[PO])
920 IF FO=1 AND T>0 THEN C1=(1-(C/T)@3)*100 ELSE C1=0
930 IF FO=2 OR FO=4 THEN C2=(1-COS(ATN(T/100))@3)*100
940 IF FO=3 OR FO=4 THEN C3=SIN(ATN(T/100)) @ C3=3*C3*C3*100
950 P1=11
960 FOR J=1 TO M
970 P2=P1+5
980 R(I,J)=VAL(O$[P1,P2])

```

```

990 IF ABS(R(I,J))<20 THEN R(I,J)=(R(I,J)+E1(J))/E0(J) @
GOTO 1010
1000 IF ABS(R(I,J))>=20 THEN R(I,J)=(R(I,J)+E2(J))/E0(J)
1010 IF F0=1 THEN R(I,J)=R(I,J)+C1
1020 IF F0=2 OR F0=4 THEN R(I,J)=R(I,J)+C2
1030 IF F0=3 OR F0=4 THEN R(I,J)=R(I,J)+C3
1040 IF FLAG(2) THEN R(I,J)=R(I,J)+100
1050 P1=P2+1
1060 P2=P1+5
1070 Q(I,J)=VAL(OS[P1,P2])
1080 IF ABS(Q(I,J))<20 THEN Q(I,J)=(Q(I,J)+E4(J))/E3(J) @
GOTO 1100
1090 IF ABS(Q(I,J))>=20 THEN Q(I,J)=(Q(I,J)+E5(J))/E3(J)
1100 P1=P2+1
1110 NEXT J
1120 NEXT I
1130 ASSIGN# 1 TO *
1140 S1=S(1) @ S2=S(N)
1150 IF NOT FLAG(3) THEN 1250
1160 FOR K=1 TO M
1170 IF F(K)=F THEN 1190
1180 NEXT K
1190 FOR I=1 TO N
1200 FOR J=1 TO M
1210 IF J=K THEN 1230
1220 R(I,J)=R(I,J)-R(I,K)
1230 NEXT J
1240 NEXT I
1250 RETURN
1260 !
1270 ! 80 CHAR/LINE PRINT
1280 !
1290 CLEAR @ CFLAG 1
1300 DISP "ENTER PRINTER ADDRESS"
1310 DISP "DEFAULT = 701" @ INPUT Z$
1320 IF Z$="" THEN P=701 @ GOTO 1370
1330 P=INT(VAL(Z$))
1340 IF P<700 OR P>799 THEN DISP "NOT AN HP-IB DEVICE"
1350 IF P<700 OR P>799 THEN DISP @ GOTO 1300
1360 !
1370 ! IMPACT PRINTER ROUTINE
1380 !
1390 CLEAR @ CRT OFF
1400 PRINTER IS P,80
1410 OUTPUT P USING "#,K" ; CHR$(27)&"C"&CHR$(72)
1420 OUTPUT P USING "#,K" ; CHR$(27)&"C"&CHR$(0)&CHR$(11)
1430 OUTPUT P USING "#,K" ; CHR$(27)&"0"
1440 OUTPUT P USING "#,K" ; CHR$(27)&"N"&CHR$(16)
1450 PRINT @ PRINT
1460 PRINT "INPUT FILE NAME = ";N$
1470 PRINT @ PRINT "SLINGRAM DATA"

```

```

1480 PRINT TRIM$(A$)&" "&TRIM$(B$)&" "&TRIM$(C$)
1490 PRINT TRIM$(D$)&" "&TRIM$(E$)
1500 PRINT TRIM$(F$)&" "&TRIM$(H$) @ PRINT
1510 T$=" "&RPT$("*****",M)
1520 PRINT T$
1530 T$=" "&RPT$(" ",M)
1540 PRINT T$
1550 T$=" "&RPT$(" FREQUENCY ",M)
1560 PRINT T$
1570 T$=""
1580 FOR I=1 TO M
1590 TO$=VAL$(F(I))
1600 L1=FLOOR((13-LEN(TO$))/2)
1610 L2=13-LEN(TO$)-L1
1620 T$=T$&RPT$(" ",L1)&TO$&RPT$(" ",L2)&" "
1630 NEXT I
1640 T$=" "&T$
1650 PRINT T$
1660 T$=" "&RPT$(" ",M)
1670 PRINT T$
1680 T$="*****"&RPT$("*****",M)
1690 PRINT T$
1700 T$=" "&RPT$(" ! ! !",M)
1710 PRINT T$
1720 T$=" STATION "&RPT$(" IP ! OP !",M)
1730 PRINT T$
1740 T$=" "&RPT$(" ! ! !",M)
1750 PRINT T$
1760 T$="!!!!!!!!!!"&RPT$("!!!!!!!!!!!!!!!!",M)
1770 PRINT T$
1780 FOR I=1 TO N
1790 T1$=""
1800 OUTPUT T1$ USING 1810 ; S(I)
1810 IMAGE #,1X,S5D.D,"!"
1820 T$=T1$
1830 FOR J=1 TO M
1840 T1$=""
1850 OUTPUT T1$ USING 1860 ; R(I,J),Q(I,J)
1860 IMAGE #,S3D.D,"!",S3D.D,"!"
1870 T$=T$&T1$
1880 NEXT J
1890 PRINT T$
1900 NEXT I
1910 PRINT CHR$(12) ! FORM-FEED
1920 PRINTER IS 2
1930 CRT ON
1940 RETURN
1950 !
1960 ! PLOTTING ROUTINE
1970 !
1980 CLEAR @ CFLAG 1

```

```

1990 DISP "ENTER PLOTTER ADDRESS"
2000 DISP "DEFAULT=708" @ INPUT Z$
2010 IF Z$="" THEN P=708 @ GOTO 2050
2020 P=INT(VAL(Z$))
2030 IF P<700 OR P>799 THEN DISP "NOT AN HP-IB DEVICE"
2040 IF P<700 OR P>799 THEN DISP @ GOTO 1990
2050 T2$=TRIM$(A$)&"/"&TRIM$(B$)&"/"&TRIM$(D$)
2060 DISP @ DISP "DEFAULT PLOT TITLE IS" @ DISP T2$ @ DISP
2070 DISP "ENTER NEW TITLE TO CHANGE(80max)"
2080 DISP "ELSE HIT ONLY END LINE" @ INPUT T3$
2090 IF T3$#"" THEN T2$=T3$
2100 DISP
2110 DISP "ENTER LIST OF FREQUENCIES"
2120 LINPUT "TO PLOT. DEFAULT IS ALL" ?",Z$
2130 IF Z$="" THEN Z$=G$
2140 P1,M0=1
2150 FOR J=1 TO M
2160 P2=POS(Z$[P1],",")+P1-2
2170 IF P2>=P1 THEN F0(J)=VAL(Z$[P1,P2])
2180 IF P2<P1 THEN F0(J)=VAL(Z$[P1]) @ GOTO 2220
2190 M0=M0+1
2200 P1=P2+2
2210 NEXT J
2220 DISP
2230 DISP "SELECT FORMAT OPTION"
2240 DISP "1=STACKED PROFILES"
2250 DISP "2=OVERLAY PROFILES"
2260 DISP "3=INDIVIDUAL PROFILES"
2270 DISP "DEFAULT=1" @ INPUT Z$
2280 IF Z$="" THEN F1=1 ELSE F1=VAL(Z$)
2290 DISP
2300 DISP "SELECT Y-AXIS SCALING OPTION"
2310 DISP "1=GROUP MAX-MIN AUTO SCALING"
2320 DISP "2=IND MAX-MIN AUTO SCALING"
2330 DISP "3=USER SELECTED SCALES"
2340 DISP "DEFAULT=1"
2350 INPUT Z$
2360 IF Z$="" THEN F2=1 ELSE F2=VAL(Z$)
2370 IF F1=2 AND F2=2 THEN F2=1 @ GOTO 2450
2380 IF F2#3 THEN 2450
2390 IF FLAG(2) THEN DISP @ DISP "ENTER IN-PHASE MIN,MAX"
2400 INPUT R1,R2
2410 IF FLAG(2) THEN DISP @ DISP "ENTER OUT-OF-PHASE
MIN,MAX"
2420 INPUT Q1,Q2
2430 IF NOT FLAG(2) THEN DISP @ DISP "ENTER GROUP MIN,MAX"
2440 INPUT G1,G2
2450 DISP
2460 DISP "SELECT X-AXIS INTERCEPT OPTION 1=CENTER OF
Y-AXIS"
2470 DISP "2=FLOATING INTERCEPT"

```

```

2480 DISP "DEFAULT=1"
2490 INPUT Z$
2500 IF Z$="" THEN F3=1 ELSE F3=VAL(Z$)
2510 DISP
2520 DISP "SELECT X-AXIS SCALING OPTION"
2530 DISP "1=MAX-MIN AUTO SCALING"
2540 DISP "2=USER SELECTED MAP SCALES          (E.G. 1:62500)"
2550 DISP "DEFAULT=1"
2560 INPUT Z$
2570 IF Z$="" THEN F4=1 ELSE F4=VAL(Z$)
2580 IF F4=1 THEN 2610
2590 IF F4=2 THEN DISP @ DISP "ENTER MAP SCALE SELECTED"
2600 INPUT U3
2610 DISP @ DISP "DISTANCE IN FEET(1) OR METERS(2)"
2620 DISP "DEFAULT=1" @ INPUT Z$
2630 IF Z$="" THEN F5=1 ELSE F5=VAL(Z$)
2640 DISP
2650 DISP "SELECT PLOTTING OPTION          1=VECTORS"
2660 DISP "2=SYMBOLS                      3=BOTH"
2670 DISP "DEFAULT=3"
2680 INPUT Z$
2690 IF Z$="" THEN F6=3 ELSE F6=VAL(Z$)
2700 DISP
2710 IF F6=1 OR F6=3 THEN DISP "ENTER SMOOTHING INTERVAL"
2720 INPUT S3@ S3=S3/S0
2730 IF FLAG(2) THEN Y0=100 ELSE Y0=0
2740 !
2750 ! Y-AXIS MAX/MIN ROUTINE
2760 !
2770 IF F2=3 THEN 3430
2780 CLEAR @ DISP "CALCULATING MAX/MIN VALUES"
2790 G1=1.E99 @ G2=-1.E99
2800 R1=1.E99 @ R2=-1.E99 @ Q1=1.E99 @ Q2=-1.E99
2810 FOR K=1 TO M0
2820 G1(K)=1.E99 @ G2(K)=-1.E99
2830 R1(K)=1.E99 @ R2(K)=-1.E99 @ Q1(K)=1.E99 @ Q2(K)=-1.E99
2840 FOR J=1 TO M
2850 IF F0(K)=F(J) THEN 2870
2860 NEXT J
2870 FOR I=1 TO N
2880 R1(K)=MIN(R1(K),R(I,J))
2890 R2(K)=MAX(R2(K),R(I,J))
2900 Q1(K)=MIN(Q1(K),Q(I,J))
2910 Q2(K)=MAX(Q2(K),Q(I,J))
2920 NEXT I
2930 R1(K)=FLOOR(R1(K)/10)*10
2940 R2(K)=CEIL(R2(K)/10)*10
2950 IF FLAG(2) THEN 2990
2960 IF R1(K)>=0 THEN R1(K)=-10
2970 IF R2(K)<=0 THEN R2(K)=10
2980 GOTO 3030

```

```

2990 IF R1(K)>=100 THEN R1(K)=90
3000 IF R2(K)<=100 THEN R2(K)=110
3010 IF F3=1 AND R2(K)-100>100-R1(K) THEN R1(K)=200-R2(K)
3020 IF F3=1 AND 100-R1(K)>R2(K)-100 THEN R2(K)=200-R1(K)
3030 Q1(K)=FLOOR(Q1(K)/10)*10
3040 Q2(K)=CEIL(Q2(K)/10)*10
3050 IF Q1(K)>=0 THEN Q1(K)=-10
3060 IF Q2(K)<=0 THEN Q2(K)=10
3070 IF FLAG(2) THEN 3130
3080 G1(K)=MIN(R1(K),Q1(K))
3090 G2(K)=MAX(R2(K),Q2(K))
3100 IF F3=1 AND G2(K)>-G1(K) THEN G1(K)=-G2(K)
3110 IF F3=1 AND -G1(K)>G2(K) THEN G2(K)=-G1(K)
3120 G1=MIN(G1,G1(K)) @ G2=MAX(G2,G2(K))
3130 R1=MIN(R1,R1(K)) @ R2=MAX(R2,R2(K))
3140 Q1=MIN(Q1,Q1(K)) @ Q2=MAX(Q2,Q2(K))
3150 NEXT K
3160 !
3170 ! CALCULATE OP MAX/MIN
3180 !
3190 IF NOT FLAG(2) THEN 3430
3200 IF F2=1 THEN 3330
3210 FOR K=1 TO M0
3220 IF Q1(K)>R1(K)-100 THEN Q1(K)=R1(K)-100
3230 IF Q1(K)<R1(K)-100 THEN R1(K)=Q1(K)+100
3240 IF Q2(K)<R2(K)-100 THEN Q2(K)=R2(K)-100
3250 IF Q2(K)>R2(K)-100 THEN R2(K)=Q2(K)+100
3260 IF F3#1 THEN 3310
3270 IF R2(K)-100>100-R1(K) THEN R1(K)=200-R2(K)
3280 IF 100-R1(K)>R2(K)-100 THEN R2(K)=200-R1(K)
3290 IF Q2(K)>-Q1(K) THEN Q1(K)=-Q2(K)
3300 IF Q2(K)<-Q1(K) THEN Q2(K)=-Q1(K)
3310 NEXT K
3320 GOTO 3430
3330 IF Q1>R1-100 THEN Q1=R1-100
3340 IF Q1<R1-100 THEN R1=Q1+100
3350 IF Q2<R2-100 THEN Q2=R2-100
3360 IF Q2>R2-100 THEN R2=Q2+100
3370 IF F3#1 THEN 3430
3380 IF R2-100>100-R1 THEN R1=200-R2
3390 IF 100-R1>R2-100 THEN R2=200-R1
3400 IF Q2>-Q1 THEN Q1=-Q2
3410 IF Q2<-Q1 THEN Q2=-Q1
3420 !
3430 ! X-AXIS MIN/MAX ROUTINE
3440 !
3450 IF F4=1 THEN X1=S1 @ X2=S2
3460 IF F4=2 AND F5=1 THEN X1=S1 @ X2=X1+148*U3/304.8/S0
3470 IF F4=2 AND F5=2 THEN X1=S1 @ X2=X1+148*U3/1000/S0
3480 X3=1
3490 IF X2-X1>=10 AND X2-X1<25 THEN X3=2

```

```

3500 IF X2-X1>=25 AND X2-X1<50 THEN X3=5
3510 IF X2-X1>=50 AND X2-X1<100 THEN X3=10
3520 IF X2-X1>=100 AND X2-X1<2500 THEN X3=25
3530 IF X2-X1>=250 AND X2-X1<5000 THEN X3=50
3540 IF X2-X1>=500 AND X2-X1<10000 THEN X3=100
3550 IF X2-X1>=1000 AND X2-X1<2500 THEN X3=250
3560 IF X2-X1>=2500 AND X2-X1<50000 THEN X3=500
3570 IF X2-X1>=5000 AND X2-X1<100000 THEN X3=1000
3580 IF X2-X1>=10000 THEN X3=5000
3590 X1=FLOOR(X1/X3)*X3
3600 IF F4=2 AND F5=1 THEN X2=X1+148*U3/304.8/S0
3610 IF F4=2 AND F5=2 THEN X2=X1+148*U3/1000/S0
3620 V1=1 ! INITIAL PAPER COUNT
3630 !
3640 ! PLOT SETUP
3650 !
3660 CLEAR
3670 DISP "LOAD PAPER AND CONTINUE" @ BEEP 200,100 @ PAUSE
3680 PLOTTER IS P
3690 LIMIT 0,265,0,185
3700 LOCATE 0,140,0,100
3710 IF F1=1 THEN D0=115/M0 ELSE D0=115
3720 FOR K=1 TO M0
3730 FOR J=1 TO M
3740 IF F0(K)=F(J) THEN 3760
3750 NEXT J
3760 IF F1<3 AND J>1 THEN 4020
3770 !
3780 ! DRAFT TITLES
3790 !
3800 SETGU
3810 MOVE 140,50 @ LDIR 270 @ LORG 6 @ CSIZE 3
3820 IF V1>1 THEN LABEL "PAGE "&VAL$(V1) @ GOTO 3990
3830 LABEL TRIM$(T2$)
3840 IF F6=2 THEN 3880
3850 IF F6=3 THEN 3900
3860 LABEL "IP = SOLID LINE      OP = DASHED LINE"
3870 GOTO 3910
3880 LABEL "IP = (*)      OP = (+)"
3890 GOTO 3910
3900 LABEL "IP = SOLID LINE AND (*)      OP = DASHED LINE AND
(+) "
3910 IF F4=2 THEN LABEL "MAP SCALE = 1:"&VAL$(U3)
3920 MOVE 3,50 @ LDIR 270 @ LORG 4
3930 IF F5=1 THEN LABEL "DISTANCE ( x "&VAL$(S0)&" ft )"
3940 IF F5=2 THEN LABEL "DISTANCE ( x "&VAL$(S0)&" m )"
3950 MOVE 70,100 @ LDIR 0 @ LORG 6
3960 IF NOT FLAG(2) THEN LABEL "PERCENT (%)" @ GOTO 4000
3970 LABEL "PERCENT (%) IN-PHASE"
3980 IF X2<S2 THEN 4000
3990 MOVE 70,0 @ LDIR 180 @ LABEL "PERCENT (%) OUT-PHASE"

```

```

4000 SETUU
4010 !
4020 ! LOCATE PLOTTING AREA
4030 !
4040 IF F1=1 THEN LOCATE (K-1)*D0+10,10+K*D0-4,10,90
4050 IF F1>1 THEN LOCATE 10,125,10,90
4060 !
4070 ! SCALE IN-PHASE PLOT AREA
4080 !
4090 IF F2#2 AND FLAG(2) THEN Y1=R1 @ Y2=R2
4100 IF F2#2 AND NOT FLAG(2) THEN Y1=G1 @ Y2=G2
4110 IF F2=2 AND FLAG(2) THEN Y1=R1(K) @ Y2=R2(K)
4120 IF F2=2 AND NOT FLAG(2) THEN Y1=G1(K) @ Y2=G2(K)
4130 SCALE Y1,Y2,X2,X1
4140 !
4150 ! LABEL EACH FREQUENCY
4160 !
4170 IF F1=2 THEN 4250
4180 SETGU
4190 LDIR 270 @ LORG 6
4200 IF F1=1 THEN MOVE 10+K*D0-1,50 ELSE MOVE 128,50
4210 LABEL F(J);" Hz"
4220 PENUP
4230 SETUU
4240 !
4250 ! DRAFT TRAVERSE AXIS
4260 !
4270 LINETYPE 1
4280 IF F1=2 AND K>1 THEN 4590
4290 MOVE Y0,X1 @ DRAW Y0,X2 @ PENUP
4300 FOR I=X1 TO X2 STEP X3
4310 MOVE Y0,I @ SETGU
4320 IMOVE -1,0 @ IDRAW 2,0
4330 PENUP @ SETUU
4340 NEXT I
4350 IF F1#3 AND K>1 THEN 4430
4360 LORG 6 @ LDIR 270
4370 L1=CEIL(X1/X3)*X3+X3
4380 L2=FLOOR(X2/X3)*X3
4390 FOR I=L1 TO L2 STEP X3
4400 MOVE Y1,I @ SETGU @ IMOVE -2,0 @ LABEL I @ SETUU
4410 NEXT I
4420 !
4430 ! DRAFT IN-PHASE AXIS
4440 !
4450 IF V1>1 THEN 4590
4460 XAXIS X1,10,Y1,Y2
4470 LDIR 270 @ LORG 8
4480 L1=CEIL(Y1/10)*10
4490 L2=FLOOR(Y2/10)*10
4500 IF (Y2-Y1)/10<=4 THEN Y3=10

```



```

4510 IF (Y2-Y1)/10>4 AND (Y2-Y1)/10<=8 THEN Y3=20
4520 IF (Y2-Y1)/10>8 THEN Y3=40
4530 IF F1#1 THEN Y3=10
4540 IF RMD(L1,Y3)#0 THEN L1=L1+ABS(RMD(L1,Y3))
4550 FOR I=L1 TO L2 STEP Y3
4560 MOVE I,X1 @ SETGU @ IMOVE 0,1 @ LABEL I @ SETUU
4570 NEXT I
4580 !
4590 ! IN-PHASE LINE PLOT
4600 !
4610 F9=0
4620 IF F6=2 THEN 4770
4630 GOSUB 5410 ! SPLINE
4640 FOR I=1 TO N
4650 IF I<N THEN S4=(S(I+1)-S(I))/S3
4660 IF I=N THEN S4=(S(I)-S(I-1))/S3
4670 FOR IO=1 TO S4
4680 Z0=S(I)+(IO-1)*S3
4690 IF Z0<S1 OR Z0>S2 THEN 4740
4700 Z1=Z0-S(I)
4710 Z2=R(I,J)+Z1*(A(I)+Z1*(B(I)+Z1*C(I)))
4720 PLOT Z2,Z0
4730 NEXT IO
4740 NEXT I
4750 PENUP
4760 !
4770 ! IN-PHASE SYMBOL PLOT
4780 !
4790 IF F6=1 THEN 4860
4800 LORG 5 @ LDIR 270
4810 FOR I=1 TO N
4820 IF S(I)<X1 OR S(I)>X2 THEN 4840
4830 MOVE R(I,J),S(I) @ LABEL "*"
4840 NEXT I
4850 !
4860 ! SCALE OUT-OF-PHASE AREA
4870 !
4880 F9=1
4890 IF NOT FLAG(2) THEN 5070
4900 IF F2#2 THEN Y1=Q1 @ Y2=Q2
4910 IF F2=2 THEN Y1=Q1(K) @ Y2=Q2(K)
4920 SCALE Y1,Y2,X2,X1
4930 IF F1=2 AND K>1 OR X2<S2 THEN 5070
4940 XAXIS X2,10,Y1,Y2
4950 LDIR 270 @ LORG 2
4960 L1=CEIL(Y1/10)*10
4970 L2=FLOOR(Y2/10)*10
4980 IF (Y2-Y1)/10<=4 THEN Y3=10
4990 IF (Y2-Y1)/10>4 AND (Y2-Y1)/10<=8 THEN Y3=20
5000 IF (Y2-Y1)/10>8 THEN Y3=40
5010 IF F1#1 THEN Y3=10

```

```

5020 IF RMD(L1,Y3)#0 THEN L1=L1+ABS(RMD(L1,Y3))
5030 FOR I=L1 TO L2 STEP Y3
5040 MOVE I,X2 @ SETGU @ IMOVE 0,-1 @ LABEL I @ SETUU
5050 NEXT I
5060 !
5070 ! OUT-OF-PHASE LINE PLOT
5080 !
5090 LINETYPE 4,2
5100 IF F6=2 THEN 5250
5110 GOSUB 5410 ! SPLINE
5120 FOR I=1 TO N
5130 IF I<N THEN S4=(S(I+1)-S(I))/S3
5140 IF I=N THEN S4=(S(I)-S(I-1))/S3
5150 FOR IO=1 TO S4
5160 Z0=S(I)+(IO-1)*S3
5170 IF Z0<S1 OR Z0>S2 THEN 5220
5180 Z1=Z0-S(I)
5190 Z2=Q(I,J)+Z1*(A(I)+Z1*(B(I)+Z1*C(I)))
5200 PLOT Z2,Z0
5210 NEXT IO
5220 NEXT I
5230 PENUP
5240 !
5250 ! OUT-OF-PHASE SYMBOL PLOT
5260 !
5270 IF F6=1 THEN 5330
5280 LORG 5 @ LDIR 270
5290 FOR I=1 TO N
5300 IF S(I)<X1 OR S(I)>X2 THEN 5320
5310 MOVE Q(I,J),S(I) @ LABEL "+"
5320 NEXT I
5330 IF F1=3 AND K<M0 THEN DISP "LOAD PAPER AND CONTINUE"
5340 IF F1=3 AND K<M0 THEN BEEP 200,100 @ PAUSE
5350 NEXT K
5360 V1=V1+1 @ X1=FLOOR(X2/X3)*X3
5370 IF X2<S2 AND F5=1 THEN X2=X1+148*U3/304.8/S0 @ GOTO
3640
5380 IF X2<S2 AND F5=2 THEN X2=X1+148*U3/1000/S0 @ GOTO 3640
5390 RETURN
5400 !
5410 ! SPLINE SUBROUTINE
5420 !
5430 IF N<2 THEN RETURN
5440 IF N<3 THEN 5860
5450 N2=N-1
5460 C(1)=S(2)-S(1)
5470 IF F9=0 THEN B(2)=(R(2,J)-R(1,J))/C(1)
5480 IF F9=1 THEN B(2)=(Q(2,J)-Q(1,J))/C(1)
5490 FOR N4=2 TO N2
5500 C(N4)=S(N4+1)-S(N4)
5510 A(N4)=2*(C(N4-1)+C(N4))

```

```

5520 IF F9=0 THEN B(N4+1)=(R(N4+1,J)-R(N4,J))/C(N4)
5530 IF F9=1 THEN B(N4+1)=(Q(N4+1,J)-Q(N4,J))/C(N4)
5540 B(N4)=B(N4+1)-B(N4)
5550 NEXT N4
5560 A(1)=-C(1)
5570 A(N)=-C(N2)
5580 B(1)=0
5590 B(N)=0
5600 IF N=3 THEN 5650
5610 B(1)=B(3)/(S(4)-S(2))-B(2)/(S(3)-S(1))
5620 B(N)=B(N2)/(S(N)-S(N-2))-B(N-2)/(S(N2)-S(N-3))
5630 B(1)=B(1)*C(1)@2/(S(4)-S(1))
5640 B(N)=-((B(N)*C(N-1)@2/(S(N)-S(N-3)))
5650 FOR W=2 TO N
5660 Z0=C(W-1)/A(W-1)
5670 A(W)=A(W)-Z0*C(W-1)
5680 B(W)=B(W)-Z0*B(W-1)
5690 NEXT W
5700 B(N)=B(N)/A(N)
5710 FOR N3=1 TO N2
5720 W=N-N3
5730 B(W)=(B(W)-C(W)*B(W+1))/A(W)
5740 NEXT N3
5750 IF F9=0 THEN
A(N)=(R(N,J)-R(N2,J))/C(N2)+C(N2)*(B(N2)+2*B(N))
5760 IF F9=1 THEN
A(N)=(Q(N,J)-Q(N2,J))/C(N2)+C(N2)*(B(N2)+2*B(N))
5770 FOR W=1 TO N2
5780 IF F9=0 THEN
A(W)=(R(W+1,J)-R(W,J))/C(W)-C(W)*(B(W+1)+2*B(W))
5790 IF F9=1 THEN
A(W)=(Q(W+1,J)-Q(W,J))/C(W)-C(W)*(B(W+1)+2*B(W))
5800 C(W)=(B(W+1)-B(W))/C(W)
5810 B(W)=3*B(W)
5820 NEXT W
5830 B(N)=3*B(N)
5840 C(N)=C(N-1)
5850 GOTO 5890
5860 IF F9=0 THEN A(1)=(R(2,J)-R(1,J))/(S(2)-S(1))
5870 IF F9=1 THEN A(1)=(Q(2,J)-Q(1,J))/(S(2)-S(1))
5880 B(1)=0 @ C(1)=0 @ A(2)=A(1) @ B(2)=0 @ C(2)=0
5890 RETURN

```